

12 Algorithmische Integration

12.1 Der Bernoulli-Algorithmus für rationale Funktionen

Im letzten Kapitel hatten wir u. a. die algorithmische Bestimmung diskreter Stammfunktionen für den speziellen Fall behandelt, daß die diskrete Stammfunktion ein hypergeometrischer Term ist.

Wir wollen uns nun in diesem Kapitel mit der Frage der algorithmischen Integration beschäftigen. Eine völlig analoge Situation zu Gospers Algorithmus erhält man, wenn man die Berechnung von Stammfunktionen *hyperexponentieller Funktionen* $f(x)$ betrachtet, für welche $f'(x)/f(x) \in \mathbb{K}(x)$ liegt, deren Stammfunktion wieder hyperexponentiell ist. Dies wurde von Almkvist und Zeilberger betrachtet [AZ1991], siehe auch [Koe1998], Kapitel 11.

Es stellt sich heraus, daß man für sogenannte *exp-log-Funktionen* – Elemente geeigneter Körpererweiterungen der rationalen Funktionen, welche Exponential- und Logarithmusfunktionen enthalten – ebenfalls einen endlichen Algorithmus zur Bestimmung einer Stammfunktion findet, welcher von Risch angegeben wurde [Ris1969]–[Ris1970]. Der *Risch-Algorithmus* wurde von vielen anderen, u. a. von Bronstein, Davenport, Rothstein und Trager vervollständigt und erweitert.

Der allgemeine Algorithmus ist recht umfangreich und vor allem algebraisch aufwendiger als im diskreten Fall, obwohl die grundlegenden Ideen denen im diskreten Fall durchaus ähnlich sind. Aufgrund eines Darstellungssatzes für die Stammfunktion, welcher bereits im 19. Jahrhundert von Liouville angegeben wurde, gelingt es – wie beim Gosper-Algorithmus – letzten Endes, einen Ansatz hinzuschreiben und durch Lösen eines linearen Gleichungssystems zum Ziel zu kommen.

Die Integration transzendenter exp-log-Funktionen (ohne die Betrachtung algebraischer Integranden) liegt in Buchform vor [Bro1997]. Dort können weitere Referenzen nachgesehen werden, und die wesentlichen Ideen für den algebraischen Fall findet man in [GCL1992].

Daher betrachten wir im vorliegenden Kapitel nur den Fall der Integration rationaler Funktionen, um die wesentlichen Ideen aufzuzeigen. Man beachte, daß auch der Gosper-Algorithmus in der Lage war, rationale Funktionen zu summieren, wenn diese wieder eine rationale oder hypergeometrische Stammfunktion besitzen. Auf

diese Weise läßt sich allerdings nicht jede rationale Funktion summieren, wie wir an den harmonischen Zahlen $H_k = \sum_k \frac{1}{k+1}$ gesehen hatten. Dies hatten wir damals nur am Rande betrachtet. Um den allgemeinen Fall der Summation beliebiger rationaler Funktionen zu Ende zu führen, hätten wir geeignete Körpererweiterungen vornehmen müssen. Dies werden wir nun im Fall der Integration rationaler Funktionen vorführen.

In fast allen Lehrbüchern der Analysis ist die folgende Methode zur Integration rationaler Funktionen angegeben, welche im Kern auf Leibniz zurückgeht, aber zuerst von Bernoulli vollständig bewiesen worden war.

Satz 12.1 Sei die rationale Funktion $r(x) = \frac{p(x)}{q(x)} \in \mathbb{R}(x)$, $p(x), q(x) \in \mathbb{R}[x]$ mit $\gcd(p(x), q(x)) = 1$ gegeben. Dann existiert eine irreduzible Faktorisierung

$$q(x) = c \cdot \prod_{i=1}^m (x - a_i)^{e_i} \cdot \prod_{j=1}^n (x^2 + b_j x + c_j)^{f_j}$$

des Nennerpolynoms über $\mathbb{R}$, wobei $c, a_i, b_j, c_j \in \mathbb{R}$ und $e_i, f_j \in \mathbb{N}$. Mit dieser *reellen Faktorisierung* gelingt eine *reelle Partialbruchzerlegung* der Form

$$r(x) = P(x) + \sum_{i=1}^m \sum_{k=1}^{e_i} \frac{A_{ik}}{(x - a_i)^k} + \sum_{j=1}^n \sum_{k=1}^{f_j} \frac{B_{jk}x + C_{jk}}{(x^2 + b_j x + c_j)^k} \quad (12.1)$$

mit $P(x) \in \mathbb{R}[x]$, $A_{ik}, B_{jk}, C_{jk} \in \mathbb{R}$ für $r(x)$, welche durch elementare Integrale gliedweise integriert werden kann (s. z. B. [Bro1997], Abschnitt 2.1). $\square$

Das Problem dieses „Algorithmus“ ist aber – und Ähnliches trifft auf viele Existenzaussagen der Mathematik zu – daß die Existenz der reellen Faktorisierung zwar gesichert ist, daß es aber keinen Algorithmus zu ihrer Bestimmung gibt.¹ Auch die Betrachtung der Nennernullstellen als algebraische Zahlen ist nicht praktikabel, da dies die Komplexität ungebührlich erhöht und vor allem das Resultat unnötig kompliziert darstellt, auch wenn dieses wie bei

$$\int \frac{x^3}{x^4 - 1} dx = \frac{1}{4} \ln(x^4 - 1) \quad (12.2)$$

gar keine algebraischen Zahlen enthält.

¹ In der Galoistheorie wird gezeigt, daß man die Nullstellen von Polynomen vom Grad 5 und höher i. a. nicht mehr durch Wurzeln darstellen kann.

Aber ein gegebener Integrand enthält ja nicht alle reellen Zahlen, sondern nur endlich viele, häufig ist der Integrand sogar ein Element von $\mathbb{Q}(x)$. Dann stellt sich aber die Frage, ob es für diesen Fall einen Algorithmus gibt, welcher eine Faktorisierung des Nennerpolynoms über $\mathbb{R}$ nicht benötigt.

Wir werden sehen, daß wir hierzu nicht einmal eine vollständige Faktorisierung des Nennerpolynoms über $\mathbb{Q}$ bzw. dem zugrundeliegenden Erweiterungskörper von $\mathbb{Q}$ benötigen, sondern daß eine quadratfreie Faktorisierung genügt. Im nächsten Abschnitt werden aber zunächst einige algebraische Vorbereitungen getroffen. Diese stellen eine algebraische Grundlage für die bei der gliedweisen Integration von (12.1) entstehenden logarithmischen Terme bereit.

12.2 Algebraische Vorbereitungen

Wie bei der Differentiation, welche wir in Abschnitt 2.7 programmiert haben sowie in Abschnitt 6.8 bei der quadratfreien Faktorisierung betrachteten, wollen wir die Integration als rein algebraische Prozedur ohne Benutzung des Grenzwertbegriffes erklären. Dies tun wir natürlich, indem wir die Integration als die Umkehrung der Differentiation betrachten. Da wir diesmal aber nicht ausschließlich mit Polynomen, sondern mit rationalen Funktionen arbeiten und auch die Arbeit mit transzendenten Funktionen innerhalb dieses Konzepts ermöglichen wollen, werden wir die in Abschnitt 6.8 eingeführten Begriffe in einem etwas allgemeineren Zusammenhang betrachten.

Definition 12.2 (Differentialkörper) Sei $\mathbb{K}$ ein Körper, und sei $D : \mathbb{K} \rightarrow \mathbb{K}$ eine Abbildung mit folgenden Eigenschaften

$$(D_1) \text{ (Summenregel) } D(f + g) = D(f) + D(g);$$

$$(D_2) \text{ (Produktregel) } D(f \cdot g) = D(f) \cdot g + f \cdot D(g).$$

Dann nennen wir D einen *Differentialoperator* oder eine *Ableitung* und $(\mathbb{K}, D)$ einen *Differentialkörper*.

Die Menge $\mathbb{K}_0 := \{c \in \mathbb{K} \mid D(c) = 0\}$ heißt *Konstantenkörper* von $(\mathbb{K}, D)$.²

Gilt für $f, g \in \mathbb{K}$ die Gleichung $D(f) = g$, dann sagen wir, f sei eine *Stammfunktion* oder ein *Integral* von g , und wir schreiben $f = \int g$.³ $\triangle$

² Im folgenden Satz wird u. a. bewiesen, daß $\mathbb{K}_0$ wirklich ein Körper ist.

³ Die dx Notation lassen wir in diesem Kapitel weg. Die Integrationsvariable ergibt sich jeweils aus dem Zusammenhang.

Man beachte, daß die Bezeichnung $\int g$ (wie in der Analysis) nicht eindeutig ist, da mit f auch jede Funktion $f + c$, $c \in \mathbb{K}_0$ wieder eine Stammfunktion ist.

Aus der Definition folgt

Satz 12.3 Jede Ableitung hat folgende wohlbekannten Eigenschaften:

(D_3) (**Konstantenregel**) $D(0) = 0$ und $D(1) = 0$;

(D_4) (**$\mathbb{K}_0$ -Linearität**) $D(af + bg) = aD(f) + bD(g)$ für alle $a, b \in \mathbb{K}_0$;

(D_5) (**Quotientenregel**) $D\left(\frac{f}{g}\right) = \frac{D(f)g - fD(g)}{g^2}$ für $g \neq 0$;

(D_6) (**Potenzregel**) $D(f^n) = n f^{n-1} \cdot D(f)$ für $n \in \mathbb{Z}$;

(D_7) (**Konstantenkörper**) Die Konstanten $\mathbb{K}_0$ bilden einen Körper;

(D_8) (**Partielle Integration**) $\int f D(g) = f \cdot g - \int D(f) g$.

Beweis: (D_3): Aus der Summenregel folgt $D(f) = D(f + 0) = D(f) + D(0)$, also $D(0) = 0$.

Sei $f \neq 0$. Dann folgt aus der Produktregel $D(f) = D(f \cdot 1) = D(f) \cdot 1 + f \cdot D(1)$, also $D(1) = 0$.

(D_4): Seien $a, b \in \mathbb{K}_0$, d. h. $D(a) = D(b) = 0$. Dann erhalten wir mit der Produktregel

$$\begin{aligned} D(af + bg) &= D(af) + D(bg) \\ &= D(a) \cdot f + a \cdot D(f) + D(b) \cdot g + b \cdot D(g) \\ &= a \cdot D(f) + b \cdot D(g). \end{aligned}$$

(D_5): Sei $g \neq 0$ gegeben. Wegen $0 = D(1) = D(g \cdot \frac{1}{g}) = D(g) \cdot \frac{1}{g} + g \cdot D(\frac{1}{g})$ ist $D(\frac{1}{g}) = -\frac{D(g)}{g^2}$. Also folgt

$$D\left(\frac{f}{g}\right) = D(f) \cdot \frac{1}{g} + f \cdot D\left(\frac{1}{g}\right) = D(f) \cdot \frac{1}{g} - f \cdot \frac{D(g)}{g^2} = \frac{D(f)g - fD(g)}{g^2},$$

welches die Behauptung liefert.

(D_6): Dies wird durch Induktion bewiesen. Hierfür beweisen wir die Aussage zunächst für $n \in \mathbb{N}_{\geq 0}$. Der Induktionsanfang für $n = 0$ ist klar. Gelte die Aussage nun für ein $n \in \mathbb{N}$. Dann folgt mit der Produktregel

$$D(f^{n+1}) = D(f \cdot f^n) = D(f) \cdot f^n + f \cdot D(f^n) = D(f) \cdot f^n + f \cdot n f^{n-1} \cdot D(f) = (n+1) f^n D(f),$$

was zu zeigen war. Für $n = -m < 0$ folgt die Aussage aber aus der Quotientenregel:

$$D\left(\frac{1}{f^m}\right) = -\frac{D(f^m)}{f^{2m}} = -\frac{m f^{m-1} D(f)}{f^{2m}} = -m f^{-m-1} D(f).$$

(D_7): Wegen (D_3) gilt $0 \in \mathbb{K}_0$ und $1 \in \mathbb{K}_0$. Aus (D_1) und (D_2) folgt weiter, daß mit $a, b \in \mathbb{K}_0$ auch $a + b \in \mathbb{K}_0$ und $a \cdot b \in \mathbb{K}_0$ liegen. Schließlich gilt wegen $0 = D(0) = D(a + (-a)) = D(a) + D(-a)$ mit $D(a) = 0$ auch $D(-a) = 0$, also impliziert $a \in \mathbb{K}_0$, daß $-a \in \mathbb{K}_0$ ist. Aus der Quotientenregel folgt schließlich, daß auch $\frac{1}{a} \in \mathbb{K}_0$ ist, falls $a \neq 0$ ist. Also ist $\mathbb{K}_0$ ein Unterkörper von $\mathbb{K}$.

(D_8): Dies folgt (wie in der Analysis) sofort aus der Produktregel. $\square$

Wir bemerken, daß das algebraische Konzept der Ableitung und der Stammfunktion auch in nullteilerfreien Ringen R mit 1 erklärt werden kann. Bei der Quotientenregel muß g dann eine Einheit sein. Die entstehende Struktur (R, D) nennt man eine *Differentialalgebra*.

Wir bemerken weiter, daß die bewiesenen Ableitungsregeln bis auf die *Kettenregel* nun vollständig sind. Die allgemeine Kettenregel kann in dem von uns betrachteten algebraischen Rahmen allerdings nicht gelten, denn zur Betrachtung allgemeiner verketteter Funktionen sind ggfs. komplizierte Körpererweiterungen notwendig. Einen Ersatz bildet die Kettenregel für die Potenzbildung (D_6).

Wir zeigen nun, daß die Ableitung in dem Körper $\mathbb{Q}(x)$ der rationalen Funktionen, welche durch die Festsetzung $D(x) = 1$ bereits eindeutig bestimmt ist, die üblichen Eigenschaften hat und damit die Ableitungsprozedur darstellt, welche uns wohl bekannt ist. Wir weisen darauf hin, daß wir in diesem Kapitel die Argumente von Polynomen der Einfachheit halber häufig weglassen.

Satz 12.4 Sei D eine Ableitung in $\mathbb{Q}(x)$ mit der Eigenschaft $D(x) = 1$.

(a) Für $r \in \mathbb{Q}$ gilt $D(r) = 0$.

(b) Für jedes Polynom $p = \sum_{k=0}^n a_k x^k \in \mathbb{Q}[x]$ gilt

$$D(p) = \sum_{k=1}^n k a_k x^{k-1}.$$

Insbesondere gilt für nichtkonstantes $p \in \mathbb{Q}[x]$ die Gradbeziehung $\deg(D(p), x) = \deg(p, x) - 1$.

(c) Der Konstantenkörper ist $\mathbb{Q}(x)_0 = \mathbb{Q}$.

(d) D ist in $\mathbb{Q}(x)$ der übliche Differentialoperator.

(e) Für jedes Polynom $p = \sum_{k=0}^n a_k x^k \in \mathbb{Q}[x]$ gilt ferner

$$\int p = \sum_{k=0}^n \frac{a_k}{k+1} x^{k+1} \in \mathbb{Q}[x] .$$

Beweis: (a): Aus $D(0) = 0$ und der Summenregel folgt durch doppelte Induktion $D(n) = 0$ für $n \in \mathbb{Z}$. Aus der Quotientenregel erhalten wir dann für $r = \frac{m}{n} \in \mathbb{Q}$ mit $m, n \in \mathbb{Z}, n \neq 0$:

$$D\left(\frac{m}{n}\right) = \frac{D(m)n - mD(n)}{n^2} = 0 .$$

(b): Zunächst folgt aus $D(x) = 1$ wegen der Potenzregel die Beziehung $D(x^k) = kx^{k-1}$. Dann bekommen wir aber für $p = \sum_{k=0}^n a_k x^k \in \mathbb{Q}[x]$ wegen $D(a_0) = 0$ aus Summen- und Produktregel⁴

$$D(p) = D\left(a_0 + \sum_{k=1}^n a_k x^k\right) = \sum_{k=1}^n D(a_k x^k) = \sum_{k=1}^n \left(D(a_k) x^k + a_k D(x^k)\right) = \sum_{k=1}^n k a_k x^{k-1} .$$

(c): Wegen (a) müssen wir nur noch zeigen, daß $D(r) = 0 \Rightarrow r \in \mathbb{Q}$ gilt. Sei also $r \in \mathbb{Q}(x)$ mit $r = p/q$ gegeben, wobei $p, q \in \mathbb{Q}[x]$ mit $q \neq 0$ und $\gcd(p, q) = 1$ gelte. Aus der Quotientenregel folgt

$$D(r) = \frac{D(p)q - pD(q)}{q^2} = 0 ,$$

also gilt $D(p)q - pD(q) = 0$ bzw.

$$D(p) = \frac{pD(q)}{q} .$$

Da nach Voraussetzung p und q keinen gemeinsamen Teiler haben, muß folglich q ein Teiler von $D(q)$ sein:

$$\frac{D(q)}{q} = s \in \mathbb{Q}[x] , \quad \text{also} \quad D(q) = s \cdot q$$

mit $\deg(s, x) \geq 0$. Ist nun $\deg(q, x) > 0$, so folgt hieraus mit der Gradbeziehung aus (b)

⁴ (b) folgt somit im wesentlichen, weil 1 und x den Polynomring $\mathbb{Q}[x]$ erzeugen. Um eine beliebige Ableitung festzulegen, genügt es also, $D(x)$ beliebig festzusetzen, s. Übungsaufgabe 12.1.

$$\deg(q, x) - 1 = \deg(s, x) + \deg(q, x) .$$

Da dies unmöglich ist, muß also $\deg(q, x) = 0$ bzw. $q \in \mathbb{Q}$ sein. Dann folgt aber aus $D(r) = 0$, daß $D(p) = 0$ ist. Aus (b) folgt hieraus aber $p \in \mathbb{Q}$, wie gewünscht.

(d): Da nach (b) Polynome die übliche Ableitung besitzen, folgt dies direkt aus der Quotientenregel.

(e): folgt direkt aus der Definition des Integrals zusammen mit (b). $\square$

Wir sehen also, daß Polynome wieder Polynome als Stammfunktionen besitzen. Es stellt sich nun die Frage – deren Antwort wir aus der Analysis bereits kennen –, ob jede Funktion $r \in \mathbb{Q}(x)$ wieder eine Stammfunktion $\int r \in \mathbb{Q}(x)$ besitzt. Daß dies nicht so ist, kann man wieder algebraisch nachweisen.

Satz 12.5 Es gibt kein $r \in \mathbb{Q}(x)$ mit $D(r) = \frac{1}{x}$. M. a. W.: Die Funktion $\frac{1}{x}$ hat keine rationale Stammfunktion.

Beweis: Wir nehmen an, es gäbe $r = p/q \in \mathbb{Q}(x)$ mit $p, q \in \mathbb{Q}[x]$, $\gcd(p, q) = 1$ derart, daß $D(\frac{p}{q}) = \frac{1}{x}$ ist. Dann gilt

$$\frac{D(p)q - pD(q)}{q^2} = \frac{1}{x}$$

bzw.

$$x(D(p)q - pD(q)) = q^2 . \quad (12.3)$$

Hieraus folgt aber, daß x ein Teiler von q^2 und daher auch ein Teiler von q ist. Es gibt also $n \in \mathbb{N}$ und $s \in \mathbb{Q}[x]$ mit $q = x^n s$ derart, daß $\gcd(s, x) = 1$ ist.⁵ Setzt man dies in (12.3) ein, so erhält man

$$x^{n+1} D(p)s - x p (n x^{n-1} s + x^n D(s)) = x^{2n} s^2 .$$

Nach Kürzen von x^n und Umsortieren erhalten wir

$$n p s = x D(p)s - x p D(s) - x^n s^2 .$$

Dies bedeutet aber, daß nun wegen $\gcd(s, x) = 1$ das Polynom p den Teiler x hat. Dies widerspricht aber der ursprünglichen Annahme der Teilerfremdheit von p und q . Damit ist gezeigt, daß es derartige p und q nicht geben kann. $\square$

Wir merken an, daß dieser Beweis dem üblichen Beweis für die Irrationalität von $\sqrt{2}$ ähnelt. Der Satz zeigt also, daß zur Integration rationaler Funktionen eine Körpererweiterung notwendig ist: Man muß Logarithmen einführen. Das Konzept der Logarithmen kann man algebraisch wie folgt erklären.

⁵ Die Zahl n ist die Ordnung der Nullstelle von q .

Definition 12.6 Sei $(\mathbb{K}, D)$ ein Differentialkörper und sei $(\mathbb{E}, D)$ ein *Differential-Erweiterungskörper* von $(\mathbb{K}, D)$, d. h. ein Erweiterungskörper von $\mathbb{K}$, welcher D nach $\mathbb{E}$ fortsetzt. Wenn es für ein gegebenes $\vartheta \in \mathbb{E} \setminus \mathbb{K}$ ein Element $u \in \mathbb{K}$ gibt, so daß

$$D(\vartheta) = \frac{D(u)}{u}$$

gilt, dann nennen wir ϑ *logarithmisch über $\mathbb{K}$* und schreiben $\vartheta = \ln u$. Der Körper $(\mathbb{E}, D)$ heißt dann eine *logarithmische Körpererweiterung* von $(\mathbb{K}, D)$. $\triangle$

Man beachte, daß die Definition logarithmischer Terme nichts anderes als die Kettenregel für Logarithmusfunktionen definiert.

Beispiel 12.7 Wir betrachten einige Integrationen, die wir zwar im Augenblick noch nicht (algebraisch) herleiten können, welche aber leicht durch Differentiation verifiziert werden können. Unser Ziel ist es, die nötigen Körpererweiterungen zu bestimmen.

(a): Offenbar ist

$$\int \frac{1}{x} = \ln x \in \mathbb{Q}(x, \ln x) .$$

Der zur Darstellung des Integrals benötigte Erweiterungskörper von $\mathbb{Q}(x)$ ist also der Körper der rationalen Funktionen in x und in dem logarithmischen Element $\ln x$. Zur Darstellung der Stammfunktion reicht also die Erweiterung von $\mathbb{Q}(x)$ um ein einziges logarithmisches Element aus.

(b):

$$\int \frac{1}{x^2 + x} = \int \left(\frac{1}{x} - \frac{1}{1+x} \right) = \ln x - \ln(1+x) \in \mathbb{Q}(x, \ln x, \ln(1+x)) .$$

Zur Darstellung dieser Stammfunktion sind zwei logarithmische Elemente nötig.

(c):

$$\int \frac{1}{x - x^3} = \int \left(\frac{1}{x} + \frac{x}{1-x^2} \right) = \ln x - \frac{1}{2} \ln(1-x^2) \in \mathbb{Q}(x, \ln x, \ln(1-x^2)) .$$

Man kann aber auch schreiben

$$\int \frac{1}{x - x^3} = \ln x - \frac{1}{2} \ln(1+x) - \frac{1}{2} \ln(1-x) \in \mathbb{Q}(x, \ln x, \ln(1+x), \ln(1-x)) .$$

In diesem Fall kommt man also, je nach Darstellung des Integrals, mit zwei bzw. drei logarithmischen Elementen aus.

(d): Bei dem Beispiel

$$\int \frac{1}{x^2 - 3} = \frac{1}{2\sqrt{3}} \ln(x - \sqrt{3}) - \frac{1}{2\sqrt{3}} \ln(x + \sqrt{3}) \in \mathbb{Q}(x)(\sqrt{3})(\ln(x - \sqrt{3}), \ln(x + \sqrt{3}))$$

muß zusätzlich zu zwei logarithmischen Termen auch noch eine algebraische Erweiterung vorgenommen werden, um die Stammfunktion darstellen zu können.

(e): Schließlich betrachten wir das Beispiel

$$I = \int \frac{x^3}{x^4 - 1}.$$

Wir hatten in (12.2) bereits gesehen, daß $I = \frac{1}{4} \ln(x^4 - 1)$ ist. Diese Darstellung des Integrals liegt also in $\mathbb{Q}(x, \ln(x^4 - 1))$. Die resultierende Darstellung ist aber abhängig vom Grad der Faktorisierung des Nenners. Eine vollständige Faktorisierung des Nenners $x^4 - 1 = (x + 1)(x - 1)(x + i)(x - i)$ über $\mathbb{Q}(x)(i)$ liefert die *Partialbruchzerlegung*

$$\frac{x^3}{x^4 - 1} = \frac{1}{4} \frac{1}{x + 1} + \frac{1}{4} \frac{1}{x - 1} + \frac{1}{4} \frac{1}{x + i} + \frac{1}{4} \frac{1}{x - i}$$

und somit die Darstellung

$$I = \frac{1}{4} \ln(x + 1) + \frac{1}{4} \ln(x - 1) + \frac{1}{4} \ln(x + i) + \frac{1}{4} \ln(x - i).$$

in $\mathbb{Q}(x)(i)(\ln(x + 1), \ln(x - 1), \ln(x + i), \ln(x - i))$. Faßt man die beiden letzten Integrale zusammen, erhält man

$$I = \frac{1}{4} \ln(x + 1) + \frac{1}{4} \ln(x - 1) + \frac{1}{4} \ln(x^2 + 1) \in \mathbb{Q}(x)(\ln(x + 1), \ln(x - 1), \ln(x^2 + 1)).$$

Der erforderliche Erweiterungskörper ist also keineswegs eindeutig bestimmt. $\triangle$

Wir bemerken noch, daß die hier resultierenden logarithmischen Terme keine Beträge enthalten, wie dies in manchen Analysisbüchern üblich ist. Dies ist auch gar nicht sinnvoll, denn über eine Singularität der Logarithmusfunktion hinweg kann man auch in der Analysis nicht integrieren. In unserem algebraischen Zugang gibt es ferner keine Zweige der Logarithmusfunktion. Eine ganz andere Sache ist aber die Auswertung von Stammfunktionen für $x \in \mathbb{Q}$. Hierfür benötigt man wieder analytische Konzepte.

Die Integration der Partialbruchzerlegung (12.1) liefert einen rationalen und einen logarithmischen Anteil. Im nächsten Abschnitt werden wir zeigen, wie man den rationalen Teil über $\mathbb{Q}(x)$ bestimmen kann, während wir uns im übernächsten Abschnitt um den logarithmischen Teil und hierbei insbesondere um die Frage, welche Körpererweiterungen zur Darstellung des Integrals nötig sind, kümmern werden.

12.3 Rationaler Teil

Sei in der Folge $\mathbb{K}$ der Einfachheit halber ein Unterkörper von $\mathbb{C}$. Will man $r(x) = \frac{p(x)}{q(x)} \in \mathbb{K}(x)$ integrieren, so besteht der erste Schritt in einer Polynomdivision, welche den polynomialen Anteil $P(x)$ abspaltet. Der Rest ist dann wieder von der Form $\frac{p(x)}{q(x)}$, wobei nun $\deg(p(x), x) < \deg(q(x), x)$ ist. Dies werden wir nun o. B. d. A. voraussetzen.

Da wir ja algebraische Erweiterungen zulassen, macht es Sinn, sich den Bernoulli-Algorithmus unter diesem Gesichtspunkt noch einmal genauer anzusehen. Die reelle Partialbruchzerlegung (12.1) kann offenbar über einem geeigneten algebraischen Erweiterungskörper durch Faktorisierung der quadratischen Nennerterme weiter zerlegt werden und erhält dann die Form

$$r(x) = \sum_{i=1}^M \sum_{k=1}^{e_i} \frac{A_{ik}}{(x - a_i)^k} \quad (12.4)$$

mit $A_{ik}, a_i \in \mathbb{C}$, wobei a_i , ($i = 1, \dots, M$) die komplexen Nullstellen des Nennerpolynoms sind und e_i die jeweilige Nullstellenordnung ist. Integration der Summanden mit $k > 1$ liefert hierbei eine rationale Funktion, während die Summanden mit $k = 1$ jeweils logarithmische Terme ergeben. Es resultiert also eine Zerlegung der Form

$$\int r(x) = \sum_{i=1}^M \frac{p_i(x)}{(x - a_i)^{e_i-1}} + \int \sum_{i=1}^M \frac{A_{i1}}{x - a_i} \quad (12.5)$$

mit $\deg(p_i(x), x) < e_i - 1$.

Zusammenfassend erhalten wir also

$$\int r(x) = \frac{a(x)}{b(x)} + \int \frac{c(x)}{d(x)}, \quad (12.6)$$

wobei $\deg(a(x), x) < \deg(b(x), x)$ und $\deg(c(x), x) < \deg(d(x), x)$ ist sowie

$$b(x) = \gcd(q(x), q'(x)) \quad (12.7)$$

(s. den Abschnitt 6.8 über quadratfreie Faktorisierung) und

$$d(x) = \frac{q(x)}{b(x)} \quad (12.8)$$

der *quadratfreie Teil* des Nenners $q(x)$ ist.

Wir werden zeigen, daß auch die Polynome $a(x), c(x) \in \mathbb{K}[x]$ sind. Gleichung (12.6) liefert also eine Zerlegung des Integrals $\int r(x)$ in einen *rationalen Teil* $\frac{a(x)}{b(x)}$ und einen *transzendenten* bzw. *logarithmischen Teil* $\int \frac{c(x)}{d(x)}$, bei welchem der Nenner $d(x)$ *quadratisch* ist. Wie man den transzendenten Teil bestimmt, werden wir im nächsten Abschnitt behandeln.

Zunächst führen wir einmal an einem Beispiel vor, wie das bisher zusammengetragene Wissen zu einem Algorithmus zur Berechnung der Darstellung (12.6) wird.

Sitzung 12.1 Wir betrachten

$$\begin{aligned} \text{In}[1] := r = & \left(3x^{16} - 19x^{15} + 43x^{14} - 20x^{13} - 91x^{12} + 183x^{11} - 81x^{10} - 166x^9 + \right. \\ & \left. 271x^8 - 101x^7 - 127x^6 + 168x^5 - 53x^4 - 31x^3 + 41x^2 - 2x - 2 \right) / \\ & \left(4x^{14} - 20x^{13} + 28x^{12} + 24x^{11} - 108x^{10} + 84x^9 + 76x^8 - \right. \\ & \left. 176x^7 + 76x^6 + 84x^5 - 108x^4 + 24x^3 + 28x^2 - 20x + 4 \right) \\ \text{Out}[1] = & (3x^{16} - 19x^{15} + 43x^{14} - 20x^{13} - 91x^{12} + 183x^{11} - 81x^{10} - 166x^9 + \\ & 271x^8 - 101x^7 - 127x^6 + 168x^5 - 53x^4 - 31x^3 + 41x^2 - 2x - 2) / \\ & (4x^{14} - 20x^{13} + 28x^{12} + 24x^{11} - 108x^{10} + 84x^9 + 76x^8 - 176x^7 + \\ & 76x^6 + 84x^5 - 108x^4 + 24x^3 + 28x^2 - 20x + 4) \end{aligned}$$

Natürlich kann *Mathematica* $r(x)$ integrieren:

$$\begin{aligned} \text{In}[2] := \text{int} = \int r dx \\ \text{Out}[2] = & \frac{x^3}{4} - \frac{x^2}{2} + \frac{x}{2} - \frac{\tan^{-1}\left(\frac{2x-1}{\sqrt{3}}\right)}{9\sqrt{3}} - \frac{1}{4} \log(x-1) - \frac{1}{12} \log(x+1) + \frac{1}{6} \log(x^2-x+1) - \\ & \frac{13}{128(x-1)} + \frac{53}{1152(x+1)} - \frac{71}{512(x-1)^2} + \frac{13}{1536(x+1)^2} + \frac{35}{192(x-1)^3} + \\ & \frac{1}{1152(x+1)^3} - \frac{15}{128(x-1)^4} + \frac{1}{40(x-1)^5} + \frac{1}{32(x-1)^6} - \frac{1}{28(x-1)^7} \end{aligned}$$

Der rationale Teil des Integrals ist also gegeben durch

$$\begin{aligned} \text{In}[3] := \text{rat} = \text{int} /. \{\text{ArcTan}[_] \rightarrow 0, \text{Log}[_] \rightarrow 0\} \\ \text{Out}[3] = & \frac{x^3}{4} - \frac{x^2}{2} + \frac{x}{2} - \frac{13}{128(x-1)} + \frac{53}{1152(x+1)} - \frac{71}{512(x-1)^2} + \frac{13}{1536(x+1)^2} + \\ & \frac{35}{192(x-1)^3} + \frac{1}{1152(x+1)^3} - \frac{15}{128(x-1)^4} + \frac{1}{40(x-1)^5} + \frac{1}{32(x-1)^6} - \frac{1}{28(x-1)^7} \end{aligned}$$

Nun berechnen wir diese Darstellung selbst. Die Eingabefunktion $r(x)$ hat den Zähler

$$\begin{aligned} \text{In}[4] := p = \text{Numerator}[r] \\ \text{Out}[4] = & 3x^{16} - 19x^{15} + 43x^{14} - 20x^{13} - 91x^{12} + 183x^{11} - 81x^{10} - \\ & 166x^9 + 271x^8 - 101x^7 - 127x^6 + 168x^5 - 53x^4 - 31x^3 + 41x^2 - 2x - 2 \end{aligned}$$

und den Nenner

In[5] := q = Denominator[r]

Out[5] = $4x^{14} - 20x^{13} + 28x^{12} + 24x^{11} - 108x^{10} + 84x^9 +$
 $76x^8 - 176x^7 + 76x^6 + 84x^5 - 108x^4 + 24x^3 + 28x^2 - 20x + 4$

Der polynomiale Anteil $P(x)$ ist gegeben durch

In[6] := Pol = PolynomialQuotient[p, q, x]

Out[6] = $\frac{3x^2}{4} - x + \frac{1}{2}$

und wir erhalten den reduzierten Zähler

In[7] := p = PolynomialRemainder[p, q, x]

Out[7] = $4x^4 + 4x^2 + 12x - 4$

Aus dem Nenner $q(x)$ können wir gemäß (12.7) und (12.8) die Nenner $b(x)$ und $d(x)$ der Zerlegung (12.6) berechnen:

In[8] := b = PolynomialGCD[q, D[q, x]]

Out[8] = $4x^{10} - 16x^9 + 12x^8 + 32x^7 - 56x^6 + 56x^4 - 32x^3 - 12x^2 + 16x - 4$

Man beachte, daß diese nun nicht in faktorisierten Form vorliegen. Dies ist auch gar nicht nötig, und vor allem ist es nicht wünschenswert, unnötige Faktorisierungen durchzuführen.

In[9] := d = Together[$\frac{q}{b}$]

Out[9] = $x^4 - x^3 + x - 1$

Wir machen nun einen Ansatz für die Zählerpolynome $a(x)$ und $c(x)$, für die wir ja jeweils eine Gradschranke besitzen:

In[10] := a = $\sum_{k=0}^{\text{Exponent}[b, x]-1} \alpha_k x^k$

Out[10] = $\alpha_9 x^9 + \alpha_8 x^8 + \alpha_7 x^7 + \alpha_6 x^6 + \alpha_5 x^5 + \alpha_4 x^4 + \alpha_3 x^3 + \alpha_2 x^2 + \alpha_1 x + \alpha_0$

In[11] := c = $\sum_{k=0}^{\text{Exponent}[d, x]-1} \beta_k x^k$

Out[11] = $\beta_3 x^3 + \beta_2 x^2 + \beta_1 x + \beta_0$

Gleichung (12.6) ist gleichwertig zu der Identität

$$p(x) - q(x) \left(\left(\frac{a(x)}{b(x)} \right)' + \frac{c(x)}{d(x)} \right) = 0.$$

Man beachte, daß dies nach Kürzen wegen (12.7) und (12.8) eine Polynomidentität ist, welche im vorliegenden Fall 91 Summanden besitzt:

In[12] := s = Together[p - q (D[$\frac{a}{b}$, x] + $\frac{c}{d}$)];

In[13] := Length[s]

Out[13] = 91

Wir führen nun einen Koeffizientenvergleich durch und lösen nach den Unbestimmten α_k und β_k auf:

```
In[14]:= sol = Solve[CoefficientList[s, x] == 0,
      Join[Table[ $\alpha_k$ , {k, 0, Exponent[b, x] - 1}],
      Table[ $\beta_k$ , {k, 0, Exponent[d, x] - 1}]]]
Out[14]= {{ $\alpha_0 \rightarrow \frac{67}{70}, \alpha_1 \rightarrow -\frac{16}{315}, \alpha_2 \rightarrow -\frac{4631}{630}, \alpha_3 \rightarrow \frac{872}{315},$ 
 $\alpha_4 \rightarrow \frac{392}{45}, \alpha_5 \rightarrow -\frac{268}{45}, \alpha_6 \rightarrow -\frac{26}{9}, \alpha_7 \rightarrow \frac{28}{9}, \alpha_8 \rightarrow -\frac{2}{9},$ 
 $\alpha_9 \rightarrow -\frac{2}{9}, \beta_0 \rightarrow \frac{1}{18}, \beta_1 \rightarrow -\frac{1}{2}, \beta_2 \rightarrow -\frac{1}{18}, \beta_3 \rightarrow 0$ }}
```

Dies liefert also den rationalen Teil $\frac{a(x)}{b(x)}$

```
In[15]:= Together[ $\frac{a}{b}$  /. sol[[1]]]
```

```
Out[15]= (-140 x^9 - 140 x^8 + 1960 x^7 - 1820 x^6 - 3752 x^5 + 5488 x^4 + 1744 x^3 - 4631 x^2 - 32 x + 603)/
(2520 (x - 1)^7 (x + 1)^3)
```

welcher natürlich mit dem von *Mathematica* berechneten übereinstimmt:

```
In[16]:= Together[rat - Integrate[Pol, x]]
```

```
Out[16]= (-140 x^9 - 140 x^8 + 1960 x^7 - 1820 x^6 - 3752 x^5 + 5488 x^4 + 1744 x^3 - 4631 x^2 - 32 x + 603)/
(2520 (x - 1)^7 (x + 1)^3)
```

sowie den transzendenten Teil $\frac{c(x)}{d(x)}$

```
In[17]:= trans = Together[ $\frac{c}{d}$  /. sol[[1]]]
```

```
Out[17]=  $\frac{-x^2 - 9x + 1}{18(x^4 - x^3 + x - 1)}$ 
```

welcher integriert

```
In[18]:= Integrate[trans, x]
```

```
Out[18]=  $\frac{1}{18} \left( -\frac{2 \tan^{-1}\left(\frac{2x-1}{\sqrt{3}}\right)}{\sqrt{3}} - \frac{9}{2} \log(x-1) - \frac{3}{2} \log(x+1) + 3 \log(x^2 - x + 1) \right)$ 
```

den rein logarithmischen Teil liefert. Dieser enthält in *Mathematica*s Ausgabe auch einen Arkustangentenstern, welcher – bei einer geeigneten algebraischen Erweiterung – auch durch komplexe Logarithmen dargestellt werden kann.

Wir können nun den gesamten Ablauf des vorliegenden Algorithmus' programmieren:

```

In[19]:= RationaleZerlegung[r_, x_] :=
Module[{p, q, Pol, a, b, c, d, s, sol, k, α, β},
  p = Numerator[r];
  q = Denominator[r];
  Pol = PolynomialQuotient[p, q, x];
  p = PolynomialRemainder[p, q, x];
  b = PolynomialGCD[q, D[q, x]];
  d = Together[q/b];
  a = Sum[α[k] x^k, {k, 0, Exponent[b, x] - 1}];
  c = Sum[β[k] x^k, {k, 0, Exponent[d, x] - 1}];
  s = Together[p - q * (D[a/b, x] + c/d)];
  sol = Solve[CoefficientList[s, x] == 0,
    Join[Table[α[k], {k, 0, Exponent[b, x] - 1}],
      Table[β[k], {k, 0, Exponent[d, x] - 1}]]];
  a = a/.sol[[1]];
  c = c/.sol[[1]];
  {Integrate[Pol, x] + Together[a/b], Together[c/d]}
]

```

Die Funktion **RationaleZerlegung** berechnet die rationale Zerlegung für beliebiges $r(x) \in \mathbb{Q}(x)$ und gibt eine Liste aus, deren erstes Element der rationale Teil $\int P(x) + \frac{a(x)}{b(x)}$ und deren zweites Element der logarithmische Teil $\frac{c(x)}{d(x)}$ ist. Für obiges Beispiel erhalten wir wie gehabt

```
In[20]:= RationaleZerlegung[r, x]
```

```

Out[20]= { $\frac{x^3}{4} - \frac{x^2}{2} + \frac{x}{2} +$ 
 $(-140x^9 - 140x^8 + 1960x^7 - 1820x^6 - 3752x^5 + 5488x^4 + 1744x^3 - 4631x^2 -$ 
 $32x + 603)/(2520(x-1)^7(x+1)^3), \frac{-x^2 - 9x + 1}{18(x^4 - x^3 + x - 1)}$ }

```

Es folgen zwei weitere Beispiele:

```
In[21]:= RationaleZerlegung[ $\frac{1}{x}$ , x]
```

```
Out[21]= {0,  $\frac{1}{x}$ }
```

```
In[22]:= RationaleZerlegung[ $\frac{x^2}{(1+x^2)^2}$ , x]
```

```
Out[22]= {- $\frac{x}{2(x^2+1)}$ ,  $\frac{1}{2(x^2+1)}$ }
```

Im ersten Fall ist der rationale Anteil gleich 0. □

Wir wollen nun zeigen, daß dieser Algorithmus, welcher zuerst von Ostrogradski [Ost1845] gefunden wurde und der damit also schon 150 Jahre alt ist, immer zum Ziel führt.

Satz 12.8 (Rationale Zerlegung) Sei $r(x) = \frac{p(x)}{q(x)} \in \mathbb{K}(x)$ eine rationale Funktion mit $p(x), q(x) \in \mathbb{K}[x]$ und $\gcd(p(x), q(x)) = 1$. Seien ferner $b(x)$ und $d(x)$ gemäß (12.7) bzw. (12.8), also durch

$$b(x) = \gcd(q(x), q'(x)) \quad \text{und} \quad d(x) = \frac{q(x)}{b(x)}$$

gegeben. Dann gibt es eindeutig bestimmte Polynome $P(x), a(x), c(x) \in \mathbb{K}[x]$ vom Grad $\deg(a(x), x) < \deg(b(x), x)$ sowie $\deg(c(x), x) < \deg(d(x), x)$ mit

$$\int r(x) = P(x) + \frac{a(x)}{b(x)} + \int \frac{c(x)}{d(x)}.$$

Das Polynom $P(x)$ erhält man durch Polynomdivision. Für $\deg(p(x), x) < \deg(q(x), x)$, also $P(x) = 0$, ist obige Gleichung gleichwertig zu der Polynomgleichung

$$p(x) - q(x) \left(\left(\frac{a(x)}{b(x)} \right)' + \frac{c(x)}{d(x)} \right) = 0. \quad (12.9)$$

Die Koeffizienten der Polynome $a(x)$ und $c(x)$ erhält man durch Ansatz mit unbestimmten Koeffizienten durch Koeffizientenvergleich aus Gleichung (12.9).

Beweis: Im ersten Schritt wird der polynomiale Anteil $P'(x)$ von $r(x)$ durch Polynomdivision abgespalten. Der Rest ist ein Polynom, dessen Grad kleiner als $\deg(q(x), x)$ ist. Wir können also in der Folge annehmen, daß $r(x) = \frac{p(x)}{q(x)} \in \mathbb{K}(x)$ mit $\deg(p(x), x) < \deg(q(x), x)$.

Seien nun $b(x) = \gcd(q(x), q'(x))$ und $d(x) = \frac{q(x)}{b(x)}$. Diese liegen offenbar in $\mathbb{K}[x]$. Wir müssen nun zeigen, daß (12.9) eine Polynomgleichung ist. Wegen $q(x) = b(x)d(x)$ ist

$$q(x) \frac{c(x)}{d(x)} = b(x) \cdot c(x) \in \mathbb{K}[x].$$

Etwas schwieriger ist es nachzuweisen, daß $q(x) \left(\frac{a(x)}{b(x)} \right)' \in \mathbb{K}[x]$ liegt. Wegen $q(x) = b(x)d(x)$ erhalten wir mit der Quotientenregel

$$q(x) \left(\frac{a(x)}{b(x)} \right)' = b(x) \cdot d(x) \left(\frac{a'(x)b(x) - a(x)b'(x)}{b(x)^2} \right) = d(x) a'(x) - a(x) \frac{d(x)b'(x)}{b(x)} \in \mathbb{K}[x],$$

falls $\frac{d(x)b'(x)}{b(x)} \in \mathbb{K}[x]$ liegt. In einem geeigneten algebraischen Erweiterungskörper

zerfällt $q(x)$ in Linearfaktoren $q(x) = c \prod_{i=1}^M (x - a_i)^{e_i}$. Für $q(x)$ sind aber nach Definition

$$b(x) = \gcd(q(x), q'(x)) = \prod_{i=1}^M (x - a_i)^{e_i - 1} \quad (\text{vgl. Abschnitt 6.8}) \quad \text{und} \quad d(x) = \frac{q(x)}{b(x)} = c \prod_{i=1}^M (x - a_i).$$

Ableiten von $b(x)$ liefert weiter $\prod_{i=1}^M (x - a_i)^{e_i - 2} \mid b'(x)$.⁶ Hieraus folgt schließlich

$$\frac{d(x)b'(x)}{b(x)} \in \mathbb{K}[x].$$

⁶ Für $e_i = 1$ gilt sogar $(x - a_i)^{e_i - 1} \mid b'(x)$.

Da nach dem Bernoullischen Satz 12.1 bzw. nach seiner komplexen Variante (12.5) genau eine Darstellung der Form (12.6) existiert, wird diese durch Lösen des resultierenden linearen Gleichungssystems auch gefunden. $\square$

12.4 Logarithmischer Teil

Um den logarithmischen Teil zu bestimmen, können wir nun also davon ausgehen, daß $r(x) = \frac{c(x)}{d(x)}$ mit $c(x), d(x) \in \mathbb{K}[x]$, wobei $d(x)$ quadratfrei und $\deg(c(x), x) < \deg(d(x), x)$ ist. Wir nehmen ferner o. B. d. A. an, daß $d(x)$ normiert ist.

Wir wissen natürlich, wie wir das Ergebnis für $\mathbb{K} = \mathbb{C}$ bzw. im kleinsten Zerfällungskörper $\mathbb{K}_{d(x)}$ von $d(x)$ über $\mathbb{K}$ erhalten: In diesem Fall hat $d(x)$ eine vollständige Faktorisierung in Linearfaktoren $d(x) = \prod_{k=1}^m (x - a_k)$, aus welcher dann eine Partialbruchzerlegung der Form

$$\frac{c(x)}{d(x)} = \sum_{k=1}^m \frac{\gamma_k}{x - a_k} \quad (\gamma_k \in \mathbb{K}_{d(x)}) \quad (12.10)$$

bzw. die Darstellung des Integrals

$$\int \frac{c(x)}{d(x)} = \sum_{k=1}^m \gamma_k \ln(x - a_k) \quad (12.11)$$

folgt. Da $d(x)$ quadratfrei ist, sind die Nullstellen a_k ($k = 1, \dots, m$) paarweise verschieden. Diese Darstellung des Integrals liegt also in dem Erweiterungskörper $\mathbb{K}_{d(x)}(x, \ln(x - a_1), \dots, \ln(x - a_m))$ von $\mathbb{K}(x)$.

Bei dem Beispiel

$$\int \frac{1}{x^3 + x} = \ln x - \frac{1}{2} \ln(x + i) - \frac{1}{2} \ln(x - i)$$

ist aber die Erweiterung von $\mathbb{Q}$ um die algebraische Zahl i beispielsweise gar nicht nötig. Da die Koeffizienten der beiden logarithmischen Terme $\ln(x \pm i)$ beide übereinstimmen, können wir nämlich das Resultat mit Hilfe der Umformungsregeln für Logarithmen vereinfachen:⁷

⁷ Wir wenden diese Umformungsregeln einfach formal an, obwohl deren Gültigkeit – insbesondere im Komplexen – nicht gesichert ist. Beim konkreten Beispiel kann man das erzielte Ergebnis dennoch durch Differentiation verifizieren. Wir werden die momentanen Vorbetrachtungen beim Beweis des folgenden Satzes aber nicht benutzen.

$$-\frac{1}{2} \ln(x+i) - \frac{1}{2} \ln(x-i) = -\frac{1}{2} \ln((x+i)(x-i)) = -\frac{1}{2} \ln(x^2+1),$$

so daß schließlich

$$\int \frac{1}{x^3+x} = \ln x - \frac{1}{2} \ln(x^2+1).$$

In dieser Form ist die algebraische Körpererweiterung um i nicht mehr nötig. Offensichtlich lassen sich auftretende Logarithmen immer dann derart zusammenfassen, wenn sie *identische* Koeffizienten haben. Also müssen wir, um unnötige Körpererweiterungen zu vermeiden, nach den *verschiedenen* Lösungen γ_k der Gleichung (12.10) suchen.

Wie werden diese Koeffizienten bestimmt? Sei a eine der Nullstellen von $d(x)$. Wegen der Quadratfreiheit von $d(x)$ sind alle Nullstellen einfach und $\frac{c(x)}{d(x)}$ hat an der Stelle a eine Laurententwicklung der Form

$$\frac{c(x)}{d(x)} = \frac{\gamma}{x-a} + p(x), \quad (12.12)$$

wobei $p(x)$ eine Potenzreihe ist.⁸ Ableiten der Identität

$$(x-a)c(x) = \gamma d(x) + (x-a)d(x)p(x)$$

erzeugt⁹

$$c(x) + (x-a)c'(x) = \gamma d'(x) + d(x)p(x) + (x-a)(d'(x)p(x) + d(x)p'(x)),$$

und Einsetzen von $x = a$ liefert wegen $d(a) = 0$ also die Formel

$$\gamma = \frac{c(a)}{d'(a)}$$

zur Berechnung von γ .¹⁰ Gleichung (12.11) kann also schließlich in der Form

$$\int \frac{c(x)}{d(x)} = \sum_{\{a \mid d(a)=0\}} \frac{c(a)}{d'(a)} \ln(x-a) \quad (12.13)$$

geschrieben werden.

⁸ Für Leser, welche die Funktionentheorie kennen: Der Koeffizient γ ist das *Residuum* von $\frac{c(x)}{d(x)}$ an der Stelle a .

⁹ Wir benutzen hierbei die Kenntnis aus der Funktionentheorie, daß die Potenzreihe $p(x)$ eine Ableitung besitzt. Man kann aber auch – ähnlich wie für Polynome – algebraisch Ableitungen von Potenzreihen erklären.

¹⁰ Integration von (12.12) zeigt, daß die gesuchten Werte γ_k genau die Residuen von $\frac{c(x)}{d(x)}$ an den Nullstellen a_k von $d(x)$ sind.

Sitzung 12.2 In komplizierten Fällen gibt *Mathematica* Integrale in eben dieser Form aus:

$$\begin{aligned} \text{In}[1] &:= \int \frac{x}{1+x+x^7} dx \\ \text{Out}[1] &= \text{RootSum}\left[\#1^7 + \#1 + 1 \&, \frac{\log(x - \#1) \#1}{7 \#1^6 + 1} \&\right] \end{aligned}$$

Zur Darstellung derartiger Summen wird also die Funktion `RootSum` verwendet.

Manchmal gelingt es *Mathematica*, die Summe weiter zu zerlegen:

$$\begin{aligned} \text{In}[2] &:= \int \frac{x}{1+x+x^5} dx \\ \text{Out}[2] &= -\frac{\tan^{-1}\left(\frac{2x+1}{\sqrt{3}}\right)}{7\sqrt{3}} + \frac{3}{14} \log(x^2 + x + 1) - \\ &\quad \frac{1}{7} \text{RootSum}\left[\#1^3 - \#1^2 + 1 \&, \frac{3 \log(x - \#1) \#1^2 - 5 \log(x - \#1) \#1 + \log(x - \#1)}{3 \#1^2 - 2 \#1} \&\right] \end{aligned}$$

Diese Zerlegung beruht auf der Faktorisierung des Nenners

$$\begin{aligned} \text{In}[3] &:= \text{Factor}[1+x+x^5] \\ \text{Out}[3] &= (x^2+x+1)(x^3-x^2+1) \end{aligned}$$

bzw. der resultierenden Partialbruchzerlegung

$$\begin{aligned} \text{In}[4] &:= \text{term} = \text{Apart}\left[\frac{x}{1+x+x^5}, x\right] \\ \text{Out}[4] &= \frac{3x+1}{7(x^2+x+1)} + \frac{-3x^2+5x-1}{7(x^3-x^2+1)} \end{aligned}$$

deren erster Summand die beiden einzelnen Summanden des Integrals liefert

$$\begin{aligned} \text{In}[5] &:= \int \text{term}[[1]] dx \\ \text{Out}[5] &= \frac{1}{7} \left(\frac{3}{2} \log(x^2+x+1) - \frac{\tan^{-1}\left(\frac{2x+1}{\sqrt{3}}\right)}{\sqrt{3}} \right) \end{aligned}$$

während der zweite Summand eine Stammfunktion hat, welche wieder durch `RootSum` dargestellt wird. $\square$

Die bisherige Methode erfordert aber immer die Berechnung aller Nullstellen a_k des Nenners $d(x)$. Wir wollen dies weitestgehend vermeiden. Unser Ziel muß es nun also sein, die *verschiedenen* Lösungen γ der Gleichung

$$0 = c(a) - \gamma d'(a)$$

zu bestimmen, wobei a die Nullstellen von $d(x)$ durchläuft, möglichst ohne explizite Berechnung der Menge $\{a \mid d(a) = 0\}$. In Abschnitt 7.5, Satz 7.30, hatten wir aber gesehen, daß dies genau die verschiedenen Nullstellen des Polynoms

$$R(z) := \text{res}(c(x) - z d'(x), d(x)) \in \mathbb{K}[z],$$

nämlich die *gemeinsamen* Nullstellen von $c(x) - z d'(x)$ und $d(x)$, sind. Auf der anderen Seite bewirkt jede *mehrfache* Nullstelle von $R(z)$, daß logarithmische Terme zusammengefaßt werden können. Wir erhalten also schließlich

$$\int \frac{c(x)}{d(x)} = \sum_{k=1}^n z_k \ln v_k(x),$$

wobei z_k die verschiedenen Nullstellen von $R(z)$ und die Polynome $v_k(x)$ normiert, quadratfrei und paarweise teilerfremd sind.

Sitzung 12.3 Wir benutzen diese Methode zur erneuten Berechnung von $\int \frac{1}{x^3+x}$:

In[1] := **c = 1; d = x³ + x;**

In[2] := **res = Resultant[c - z D[d, x], d, x]**

Out[2] = $-4z^3 + 3z + 1$

Diese Resultante stellt die Koeffizienten der Partialbruchzerlegung bereit. Ihre Faktorisierung

In[3] := **Factor[res]**

Out[3] = $-(z-1)(2z+1)^2$

zeigt, daß diese Koeffizienten $\gamma_1 = 1$ sowie $\gamma_2 = -\frac{1}{2}$ sind, wobei γ_2 eine doppelte Nullstelle ist, welche also zwei logarithmische Terme zusammenfaßt.

Im nächsten Satz wird angegeben, wie sich hieraus auch die zugehörigen Funktionen $v_k(x)$ bestimmen lassen: Die Rechnung

In[4] := **PolynomialGCD[c - z D[d, x] /. z -> 1, d]**

Out[4] = x

liefert $v_1(x) = x$ und die Rechnung

In[5] := **PolynomialGCD[c - z D[d, x] /. z -> - $\frac{1}{2}$, d]**

Out[5] = $\frac{x^2}{2} + \frac{1}{2}$

zeigt $v_2(x) = x^2 + 1$, wenn wir $v_k(x)$ normiert wählen. □

Im folgenden Satz, welcher auf Rothstein [Rot1976] und Trager [Tra1976] zurückgeht, wird gezeigt, daß diese Methode immer eine Darstellung des gesuchten Integrals liefert und daß alle verschiedenen Nullstellen von $R(z)$ auch wirklich nötig sind. Ferner liefert er einen Algorithmus zur Bestimmung der Polynome $v_k(x)$.

Satz 12.9 Sei $\mathbb{E}(x)$ ein Differentialkörper über einem Konstantenkörper $\mathbb{E}$ und seien $c(x), d(x) \in \mathbb{E}[x]$ gegeben mit $\gcd(c(x), d(x)) = 1$, $d(x)$ normiert und quadratfrei sowie $\deg(c(x), x) < \deg(d(x), x)$. Gelte ferner

$$\int \frac{c(x)}{d(x)} = \sum_{k=1}^n z_k \ln v_k(x), \quad (12.14)$$

wobei $z_k \in \mathbb{E} \setminus \{0\}$ ($k = 1, \dots, n$) untereinander verschiedene Konstanten und $v_k(x) \in \mathbb{E}[x]$ ($k = 1, \dots, n$) normierte und quadratfreie, paarweise teilerfremde Polynome mit positivem Grad seien.

Dann sind die Zahlen z_k die verschiedenen Lösungen der Polynomgleichung

$$R(z) = \text{res}(c(x) - z d'(x), d(x)) = 0$$

und $v_k(x)$ sind die Polynome

$$v_k(x) = \gcd(c(x) - z_k d'(x), d(x)) \in \mathbb{E}[x].$$

Beweis: Wir differenzieren die Gleichung (12.14) und erhalten

$$\frac{c(x)}{d(x)} = \sum_{k=1}^n z_k \frac{v'_k(x)}{v_k(x)}.$$

Setzt man

$$u_k(x) := \frac{\prod_{j=1}^n v_j(x)}{v_k(x)} = \prod_{\substack{j=1 \\ j \neq k}}^n v_j(x) \in \mathbb{E}(x),$$

so liefert dies nach Multiplikation mit $d(x) \cdot \prod_{j=1}^n v_j(x)$

$$c(x) \prod_{j=1}^n v_j(x) = d(x) \sum_{k=1}^n z_k v'_k(x) u_k(x). \quad (12.15)$$

Wir behaupten nun zunächst, daß $d(x) = \prod_{j=1}^n v_j(x)$ ist. Wegen $\gcd(c(x), d(x)) = 1$ folgt aus (12.15), daß $d(x) \mid \prod_{j=1}^n v_j(x)$. Um die umgekehrte Teilbarkeitsrelation zu beweisen, stellen wir zunächst fest, daß (12.15) für jedes $j = 1, \dots, n$ die Beziehung

$$v_j(x) \mid d(x) \sum_{k=1}^n z_k v'_k(x) u_k(x)$$

impliziert. Da $v_j(x)$ für $k \neq j$ aber ein Faktor von $u_k(x)$ ist, folgt hieraus für den Summanden mit $k = j$

$$v_j(x) \mid d(x) v_j'(x) u_j(x) .$$

Nun ist aber weiter $\gcd(v_j(x), v_j'(x)) = 1$, da $v_j(x)$ quadratfrei ist, und ferner ist $\gcd(v_j(x), u_j(x)) = 1$, da die Polynome $v_j(x)$ als paarweise teilerfremd vorausgesetzt waren. Daher bleibt für alle $j = 1, \dots, n$ die Konklusion $v_j(x) \mid d(x)$, also wieder aufgrund der Teilerfremdheit schließlich $\prod_{j=1}^n v_j(x) \mid d(x)$. Da alle beteiligten Polynome normiert sind, ist hiermit gezeigt, daß

$$d(x) = \prod_{k=1}^n v_k(x) \quad (12.16)$$

gilt. Also folgt aus (12.15) weiter die Beziehung

$$c(x) = \sum_{k=1}^n z_k v_k'(x) u_k(x) .$$

Differentiation von (12.16) liefert

$$d'(x) = \sum_{k=1}^n v_k'(x) u_k(x) ,$$

also erhalten wir

$$c(x) - z_j d'(x) = \sum_{k=1}^n z_k v_k'(x) u_k(x) - z_j \sum_{k=1}^n v_k'(x) u_k(x) = \sum_{k=1}^n (z_k - z_j) v_k'(x) u_k(x) . \quad (12.17)$$

Da in dieser Summe für $j \neq k$ der Faktor $v_j(x) \mid u_k(x)$ und für $j = k$ der Summand verschwindet, gilt also für alle $j = 1, \dots, n$

$$v_j(x) \mid c(x) - z_j d'(x) . \quad (12.18)$$

Aus (12.16) und (12.18) folgt nun, daß $v_j(x)$ ein *gemeinsamer* Teiler von $d(x)$ und von $c(x) - z_j d'(x)$ ist. Wir wollen schließlich zeigen, daß $v_j(x)$ der *größte* gemeinsame Teiler dieser beiden Polynome ist. Hierfür genügt es wegen (12.16) zu zeigen, daß für $j \neq i$ die Beziehung $\gcd(c(x) - z_j d'(x), v_i(x)) = 1$ gültig ist.

Wir erhalten

$$\begin{aligned} \gcd(c(x) - z_j d'(x), v_i(x)) &= \gcd\left(\sum_{k=1}^n (z_k - z_j) v_k'(x) u_k(x), v_i(x)\right) \\ &= \gcd((z_i - z_j) v_i'(x) u_i(x), v_i(x)) , \end{aligned}$$

wobei die letzte Gleichheit folgt, da $v_i(x)$ für jedes $j \neq i$ ein Faktor von $u_j(x)$ ist. Für $j \neq i$ ist der letzte Term aber wegen $z_i \neq z_j$ (z_k sind paarweise verschieden), $\gcd(v_i'(x), v_i(x)) = 1$ (v_k sind quadratfrei) und $\gcd(u_i(x), v_i(x)) = 1$ (v_k sind paarweise verschieden) gleich 1. Also haben wir bewiesen, daß

$$v_j(x) = \gcd(c(x) - z_j d'(x), d(x)) \quad \text{für alle } j = 1, \dots, n.$$

Demnach haben $c(x) - z_j d'(x)$ und $d(x)$ einen nichttrivialen gemeinsamen Faktor, und folglich ist

$$\text{res}(c(x) - z_j d'(x), d(x)) = 0.$$

Damit haben wir zwar die Gleichungen des Satzes gezeigt, es fehlt aber noch der Nachweis, daß wirklich alle Nullstellen von $R(z)$ in (12.14) benötigt werden.

Sei also z eine beliebige Nullstelle von $R(z)$.¹¹ Dann ist $\text{res}(c(x) - z d'(x), d(x)) = 0$ und daher

$$\gcd(c(x) - z d'(x), d(x)) = g(x)$$

mit $\deg(g(x), x) > 0$. Insbesondere ist $z \neq 0$, da $c(x)$ und $d(x)$ relativ prim sind.

Ist nun $h(x)$ ein irreduzibler Faktor von $g(x)$, dann ist also $h(x) \mid d(x)$, und wegen (12.16) gibt es genau ein $j = 1, \dots, n$ mit $h(x) \mid v_j(x)$. Weiter ist $h(x) \mid c(x) - z d'(x)$, also wegen (12.17)

$$h(x) \mid \sum_{k=1}^n (z_k - z) v_k'(x) u_k(x).$$

Weil $h(x) \mid v_j(x) \mid u_k(x)$ für jedes $j \neq k$ ist, folgt also

$$h(x) \mid (z_j - z) v_j'(x) u_j(x).$$

Da $h(x)$ als Teiler von $v_j(x)$ weder ein Teiler von $v_j'(x)$ noch von $u_j(x)$ sein kann, muß also $z = z_j$ sein. Daher haben wir gezeigt, daß die willkürlich gewählte Nullstelle z von $R(z)$ eine der gegebenen Zahlen $z_j \in \mathbb{E}$ sein muß. Dies zeigt aber auch, daß die gegebenen Zahlen z_j ($j = 1, \dots, n$) genau die verschiedenen Nullstellen von $R(z)$ sind. $\square$

Der Nachteil von Satz 12.9 ist natürlich, daß wir a priori eine Integraldarstellung der gewünschten Form gefordert hatten. Diese steht aber i. a. nur in einem geeigneten algebraischen Erweiterungskörper zur Verfügung. Daher hatten wir den Differentialkörper in Satz 12.9 mit $\mathbb{E}$ bezeichnet. Wir wollen nun zeigen, daß der Algorithmus von Rothstein und Trager die *kleinstmögliche* algebraische Erweiterung erfordert. Bevor wir dies tun, betrachten wir allerdings einige Beispiele mit *Mathematica*.

¹¹ Im allgemeinen könnte $z \in \mathbb{E}_R$ im Zerfällungskörper bzgl. $R(z)$ liegen. Wir werden aber sehen, daß $z = z_k$ gelten muß, und da wir $z_k \in \mathbb{E}$ angenommen hatten, spielen sich doch alle Rechnungen in $\mathbb{E}$ ab.

Sitzung 12.4 Wir betrachten nochmals unser Beispiel $\int \frac{1}{x^3+x}$ und vervollständigen die Rechnungen aus Sitzung 12.3. Nach Eingabe von Zähler und Nenner

```
In[1]:= c = 1; d = x3 + x;
```

berechnen wir die Resultante $R(z)$

```
In[2]:= res = Resultant[c - z D[d, x], d, x]
```

```
Out[2]= -4 z3 + 3 z + 1
```

und suchen ihre Nullstellen mit `Solve`:

```
In[3]:= sol = Solve[res == 0, z]
```

```
Out[3]= {{z -> -1/2}, {z -> -1/2}, {z -> 1}}
```

Da eine der Nullstellen *mehrfach* auftritt, wir aber nur an den *verschiedenen* Nullstellen interessiert sind, entfernen wir doppelte Nullstellen mit `Union`:

```
In[4]:= sol = Union[Solve[res == 0, z]]
```

```
Out[4]= {{z -> -1/2}, {z -> 1}}
```

Dies liefert die Liste der Werte z_k

```
In[5]:= zlist = z /. sol
```

```
Out[5]= {-1/2, 1}
```

als auch die Liste der Polynome $v_k(x)$:¹²

```
In[6]:= vlist = PolynomialGCD[c - z D[d, x] /. sol, d]
```

```
Out[6]= {x2/2 + 1/2, x}
```

Mathematica gibt diese nicht normiert aus, daher normieren wir sie mit der Hilfsfunktion¹³

```
In[7]:= makemonomic[p_, x_] := Module[{n},
    n = Exponent[p, x];
    Expand[p/Coefficient[p, x, n]]
]
```

```
In[8]:= makemonomic[vlist, x]
```

```
Out[8]= {x2 + 1, x}
```

Das gesuchte Integral erhalten wir schließlich durch ein *Skalarprodukt* $z \cdot \ln v(x) = \sum_{k=1}^n z_k \ln v_k(x)$:

```
In[9]:= zlist . Log[makemonomic[vlist, x]]
```

¹² In den folgenden Befehlen wird ausgenutzt, daß viele *Mathematica*-Funktionen das Attribut `Listable` haben – betrachte beispielsweise `PolynomialGCD` – und sich daher (auch ohne `Map`) auf Listen anwenden lassen.

¹³ Dies ist nicht unbedingt erforderlich, ohne diesen Schritt erhalten wir lediglich eine andere Integrationskonstante. Wir wollen aber die Ausgabe standardisieren.

$$\text{Out}[9] = \log(x) - \frac{1}{2} \log(x^2 + 1)$$

Wir betrachten ein komplizierteres Beispiel, welches wir aus einer gewünschten Integraldarstellung konstruieren. Das Integral sei gegeben durch

$$\text{In}[10] := \text{int} = \frac{1}{2} \text{Log}[x^3 + x - 1] - \frac{2}{3} \text{Log}[x^5 + 2x^2 - 3]$$

$$\text{Out}[10] = \frac{1}{2} \log(x^3 + x - 1) - \frac{2}{3} \log(x^5 + 2x^2 - 3)$$

Dies liefert also die rationale Eingabefunktion

$$\text{In}[11] := \mathbf{r} = \mathbf{Together}[\mathbf{D}[\text{int}, \mathbf{x}]]$$

$$\text{Out}[11] = \frac{-11x^7 - 17x^5 + 22x^4 - 37x^2 + 16x - 9}{6(x^3 + x - 1)(x^5 + 2x^2 - 3)}$$

Wir berechnen Zähler und Nennerpolynom:

$$\text{In}[12] := \mathbf{c} = \frac{\mathbf{Numerator}[\mathbf{r}]}{6}; \mathbf{d} = \frac{\mathbf{Denominator}[\mathbf{r}]}{6};$$

unter Beachtung, daß $d(x)$ normiert ist. Die Resultante ergibt sich zu

$$\text{In}[13] := \mathbf{res} = \mathbf{Resultant}[\mathbf{c} - \mathbf{z D}[\mathbf{d}, \mathbf{x}], \mathbf{d}, \mathbf{x}]$$

$$\text{Out}[13] = -\frac{499499001}{8} (8z^3 - 12z^2 + 6z - 1) (243z^5 + 810z^4 + 1080z^3 + 720z^2 + 240z + 32)$$

mit den verschiedenen Nullstellen

$$\text{In}[14] := \mathbf{sol} = \mathbf{Union}[\mathbf{Solve}[\mathbf{res} == \mathbf{0}, \mathbf{z}]]$$

$$\text{Out}[14] = \left\{ \left\{ z \rightarrow -\frac{2}{3} \right\}, \left\{ z \rightarrow \frac{1}{2} \right\} \right\}$$

Also haben wir die Koeffizienten

$$\text{In}[15] := \mathbf{zlist} = \mathbf{z} /. \mathbf{sol}$$

$$\text{Out}[15] = \left\{ -\frac{2}{3}, \frac{1}{2} \right\}$$

und die Logarithmenargumente

$$\text{In}[16] := \mathbf{vlist} = \mathbf{PolynomialGCD}[\mathbf{c} - \mathbf{z D}[\mathbf{d}, \mathbf{x}] /. \mathbf{sol}, \mathbf{d}]$$

$$\text{Out}[16] = \left\{ \frac{x^5}{6} + \frac{x^2}{3} - \frac{1}{2}, \frac{x^3}{6} + \frac{x}{6} - \frac{1}{6} \right\}$$

$$\text{In}[17] := \mathbf{makemonie}[\mathbf{vlist}, \mathbf{x}]$$

$$\text{Out}[17] = \{x^5 + 2x^2 - 3, x^3 + x - 1\}$$

und somit das Lösungsintegral

$$\text{In}[18] := \mathbf{zlist} . \mathbf{Log}[\mathbf{makemonie}[\mathbf{vlist}, \mathbf{x}]]$$

$$\text{Out}[18] = \frac{1}{2} \log(x^3 + x - 1) - \frac{2}{3} \log(x^5 + 2x^2 - 3)$$

Mathematica's `Integrate`-Kommado liefert dieselbe Lösung in geringfügig anderer Darstellung:

$$\text{In}[19] := \int \mathbf{r} \, \mathbf{dx}$$

$$\text{Out}[19] = \frac{1}{6} (3 \log(x^3 + x - 1) - 4 \log(x^5 + 2x^2 - 3))$$

Natürlich ist diese Methode nur dann besser als die in Sitzung 12.2 dargestellte, falls das Polynom $R(z)$ sich in vernünftiger Weise faktorisieren läßt. Bei dem dortigen Beispiel

$$\text{In}[20] := \mathbf{c} = \mathbf{x}; \mathbf{d} = \mathbf{1} + \mathbf{x} + \mathbf{x}^7$$

erhalten wir für die Resultante

$$\text{In}[21] := \mathbf{res} = \mathbf{Resultant}[\mathbf{c} - \mathbf{z} \mathbf{D}[\mathbf{d}, \mathbf{x}], \mathbf{d}, \mathbf{x}]$$

$$\text{Out}[21] = -870199 z^7 + 37044 z^5 - 9604 z^4 - z - 1$$

welche sich nicht über $\mathbb{Q}$ faktorisieren läßt:

$$\text{In}[22] := \mathbf{Factor}[\mathbf{res}]$$

$$\text{Out}[22] = -870199 z^7 + 37044 z^5 - 9604 z^4 - z - 1$$

In diesem Fall ist die Darstellung (12.13) dem Rothstein-Trager-Algorithmus vorzuziehen, denn dieser liefert ebenfalls eine Summe über die (unbekannten) Nullstellen eines Polynoms.

Wir können nun die betrachteten Rechenschritte des Rothstein-Trager-Algorithmus' in folgender *Mathematica*-Funktion zusammenfassen:

```
In[23] := Clear[RothsteinTrager];
RothsteinTrager[r_, x_] :=
Module[{c, d, dmonic, z, res, sol, zlist, vlist},
  c = Numerator[r];
  d = Denominator[r];
  dmonic = makemonomic[d, x];
  c = Together[c * dmonic/d];
  d = dmonic;
  res = Resultant[c - z * D[d, x], d, x];
  sol = Union[Solve[res == 0, z]];
  zlist = z/.sol;
  vlist = PolynomialGCD[c - z * D[d, x]/.sol, d, Extension -> Automatic];
  zlist . Log[makemonomic[vlist, x]]
]
```

Die eben betrachteten Beispiele können nun in einem Schritt berechnet werden:

$$\text{In}[24] := \mathbf{RothsteinTrager}\left[\frac{1}{x^3 + x}, \mathbf{x}\right]$$

$$\text{Out}[24] = \log(x) - \frac{1}{2} \log(x^2 + 1)$$

$$\text{In}[25] := \mathbf{RothsteinTrager}[\mathbf{r}, \mathbf{x}]$$

$$\text{Out}[25] = \frac{1}{2} \log(x^3 + x - 1) - \frac{2}{3} \log(x^5 + 2x^2 - 3)$$

Bei dem Beispiel

$$\text{In}[26] := \text{RothsteinTrager}\left[\frac{1}{x^2 - 2}, x\right]$$

$$\text{Out}[26] = \frac{\log(x - \sqrt{2})}{2\sqrt{2}} - \frac{\log(x + \sqrt{2})}{2\sqrt{2}}$$

ist $R(z)$ nicht über $\mathbb{Q}$ faktorisierbar und liefert somit automatisch die nötige algebraische Körpererweiterung um $\sqrt{2}$. Die korrekte Berechnung des größten gemeinsamen Teilers in $\mathbb{Q}(\sqrt{2})[x]$ wurde in der Implementierung durch die Option `Extension→Automatic` bewirkt. Man beachte, daß *Mathematica* dieses Integral durch eine hyperbolische Arkustangensfunktion darstellt:

$$\text{In}[27] := \int \frac{1}{x^2 - 2} dx$$

$$\text{Out}[27] = -\frac{\tanh^{-1}\left(\frac{x}{\sqrt{2}}\right)}{\sqrt{2}}$$

für welche allerdings dieselbe algebraische Erweiterung nötig ist.

Wir wollen nun das Beispiel aus Sitzung 12.1 fortführen. Dort hatte sich als logarithmischer Teil die Funktion

$$\frac{c(x)}{d(x)} = \frac{-x^2 - 9x + 1}{18(x^4 - x^3 + x - 1)}$$

ergeben. Wir erhalten nun

$$\text{In}[28] := \text{RT} = \text{RothsteinTrager}\left[\frac{-x^2 - 9x + 1}{18(x^4 - x^3 + x - 1)}, x\right]$$

$$\text{Out}[28] = -\frac{1}{4} \log(x - 1) - \frac{1}{12} \log(x + 1) +$$

$$\frac{1}{54} (9 + i\sqrt{3}) \log\left(x - \frac{i\sqrt{3}}{2} - \frac{1}{2}\right) + \frac{1}{54} (9 - i\sqrt{3}) \log\left(x + \frac{i\sqrt{3}}{2} - \frac{1}{2}\right)$$


eine Darstellung in $\mathbb{Q}(x)(\sqrt{3}, i)(\ln(x - 1), \ln(x + 1), \ln(x - \frac{i\sqrt{3}}{2} - \frac{1}{2}), \ln(x + \frac{i\sqrt{3}}{2} - \frac{1}{2}))$, welche insbesondere *komplex* ist, obwohl es auch eine *reelle* Darstellung des Integrals gibt:

$$\text{In}[29] := \text{int} = \int \frac{-x^2 - 9x + 1}{18(x^4 - x^3 + x - 1)} dx$$

$$\text{Out}[29] = \frac{1}{18} \left(-\frac{2 \tan^{-1}\left(\frac{2x-1}{\sqrt{3}}\right)}{\sqrt{3}} - \frac{9}{2} \log(x - 1) - \frac{3}{2} \log(x + 1) + 3 \log(x^2 - x + 1) \right)$$

Wir kommen bald darauf zurück, wie man diese erhält. Wir wollen nun testen, wie sich die beiden verschiedenen Stammfunktionen unterscheiden. Es folgt eine graphische Darstellung der reellen Lösung


$$\text{In}[30] := \text{Plot}[\text{int}, \{x, 1, 3\}]$$



Out[30]= -Graphics-

sowie eine graphische Darstellung der Rothstein-Trager-Lösung


In[31]:= **Plot**[RT, {x, 1, 3}]



Out[31]= -Graphics-

welche sich ganz offensichtlich um eine Integrationskonstante (der Größe 0.100767) unterscheiden:

In[32]:= **Plot**[RT - int, {x, 1, 3}]



`Out[32]= -Graphics-`

Das Zittern bei der letzten Graphik beruht auf der numerischen Rechenungenauigkeit.¹⁴ Wir wollen diese Differenz nun berechnen. Zunächst überprüfen wir, daß tatsächlich die Differenz der beiden Integrale konstant ist:

`In[33]:= D[RT - int, x]//Together`

`Out[33]= 0`

Die Differenz kann also beispielsweise durch Einsetzen von $x = 0$ bestimmt werden:

`In[34]:= const = RT - int /. x -> 0`

$$\begin{aligned} \text{Out[34]} = & -\frac{i\pi}{4} + \frac{1}{18} \left(\frac{9i\pi}{2} - \frac{\pi}{3\sqrt{3}} \right) + \\ & \frac{1}{54} (9 + i\sqrt{3}) \log \left(-\frac{1}{2} - \frac{i\sqrt{3}}{2} \right) + \frac{1}{54} (9 - i\sqrt{3}) \log \left(-\frac{1}{2} + \frac{i\sqrt{3}}{2} \right) \end{aligned}$$

Zur Vereinfachung verwenden wir `ComplexExpand`:

`In[35]:= const = ComplexExpand[const]`

$$\text{Out[35]} = \frac{\pi}{18\sqrt{3}}$$

`In[36]:= N[const]`

`Out[36]= 0.100767`

Mathematica ist nicht perfekt beim Vereinfachen komplexer Funktionen, und es gehört einiges Geschick dazu, die bestmögliche Vereinfachung zu finden. Es ist uns beispielsweise nicht gelungen, die Differenz `RT - int` direkt zu vereinfachen. $\square$

Wir zeigen nun, daß der Rothstein-Trager-Algorithmus die kleinstmögliche algebraische Erweiterung erzeugt.

¹⁴ Die man natürlich bei der Option `PlotRange -> {0, 1}` nicht mehr sieht.

Satz 12.10 Sei $\mathbb{K}(x)$ ein Differentialkörper über einem Konstantenkörper $\mathbb{K}$. Seien weiter $c(x), d(x) \in \mathbb{K}[x]$ mit $\gcd(c(x), d(x)) = 1$, $d(x)$ normiert und quadratfrei sowie $\deg(c(x), x) < \deg(d(x), x)$. Sei schließlich $\mathbb{E}$ der kleinste algebraische Erweiterungskörper von $\mathbb{K}$ mit der Eigenschaft, daß das Integral in der Form

$$\int \frac{c(x)}{d(x)} = \sum_{k=1}^N Z_k \ln V_k(x)$$

dargestellt werden kann, wobei $Z_k \in \mathbb{E}$ und $V_k(x) \in \mathbb{E}[x]$ liegen. Dann ist

$$\mathbb{E} = \mathbb{K}(z_1, \dots, z_n),$$

wobei $z_1, \dots, z_n$ die verschiedenen Nullstellen des Polynoms

$$R(z) = \text{res}(c(x) - z d'(x), d(x)) \in \mathbb{K}[z]$$

sind, d. h., $\mathbb{E}$ ist der Zerfällungskörper von $\mathbb{K}$ bzgl. $R(z)$. Es gilt dann wieder

$$\int \frac{c(x)}{d(x)} = \sum_{k=1}^n z_k \ln v_k(x) \quad (12.19)$$

und

$$v_k(x) = \gcd(c(x) - z_k d'(x), d(x)) \in \mathbb{E}[x].$$

Beweis: Falls die Polynome $V_k(x)$ nicht die Voraussetzungen aus Satz 12.9 erfüllen (normiert, quadratfrei, paarweise verschieden, teilerfremd, positiver Grad), kann dies durch Umarrangieren mit den Logarithmengesetzen erreicht werden (s. [GCL1992], Theorem 11.8). Nach dieser Umformung liegt eine Darstellung derselben Form vor, die die Voraussetzungen von Satz 12.9 erfüllt und welche durch (12.18) (oder (12.19)) gegeben sei.

Satz 12.9 kann dann angewendet werden und besagt, daß die Werte z_k ($k = 1, \dots, n$) die verschiedenen Nullstellen des Resultantenpolynoms $R(z)$ sind. Da $R(z)$ per Definition in $\mathbb{K}[z]$ liegt, ist also $\mathbb{E}$ der Zerfällungskörper von $R(z)$. Da in Satz 12.9 gezeigt worden war, daß alle Nullstellen von $R(z)$ für die Darstellung (12.18) benötigt werden, ist $\mathbb{E}$ der kleinstmögliche algebraische Erweiterungskörper. $\square$

Zum Abschluß wollen wir zeigen, wie man (im Falle reeller Integranden) auf komplexe logarithmische Darstellungen verzichten kann. Hierzu werden die auftretenden Logarithmen in Arkustangensfunktionen konvertiert. Eine Konversion in hyperbolische Funktionen, welche *Mathematica* in einzelnen Fällen durchführt, wie wir gesehen haben, werden wir hingegen nicht betrachten.

Zunächst erklären wir in naheliegender Weise, wie Arkustangensterme algebraisch gegeben sind.

Definition 12.11 Sei $(\mathbb{K}, D)$ ein Differentialkörper und sei $(\mathbb{E}, D)$ ein Differential-Erweiterungskörper von $(\mathbb{K}, D)$. Wenn es für ein gegebenes $\varphi \in \mathbb{E} \setminus \mathbb{K}$ ein Element $u \in \mathbb{K}$ gibt, so daß

$$D(\varphi) = \frac{D(u)}{1 + u^2}$$

gilt, dann nennen wir φ *einen Arkustangens über $\mathbb{K}$* und schreiben $\varphi = \arctan u$. $\triangle$

Wieder haben wir auf diese Weise also die Kettenregel erklärt, diesmal für die Arkustangensfunktion.

Falls das Nennerpolynom $d(x) \in \mathbb{R}[x]$ komplexe Nullstellen besitzt, treten diese bekanntlich immer als Paare konjugiert komplexer Zahlen auf, s. Übungsaufgabe 12.3. Also treten beim Integrieren von $\frac{c(x)}{d(x)}$ immer Logarithmenpaare mit komplexen Argumenten auf, und man sieht leicht ein, daß die zugehörigen Koeffizienten ebenfalls konjugiert zueinander sind. Die Logarithmen mit reellen Koeffizienten lassen sich sofort mit Hilfe der Gleichung

$$c \ln(a + i b) + c \ln(a - i b) = c \ln((a + i b)(a - i b)) = c \ln(a^2 + b^2)$$

durch reelle Logarithmen ausdrücken. Wir untersuchen nun, wie wir die Logarithmen mit rein komplexen Koeffizienten zu reellen Arkustangenstermen zusammenfassen können.

Sei allgemein $u(x) \in \mathbb{K}(x)$ mit $u(x)^2 \neq -1$. Dann gilt¹⁵

$$i D(\ln(u(x) + i) - \ln(u(x) - i)) = 2 D \arctan u(x). \quad (12.20)$$


Dies folgt aus der Rechnung

$$\begin{aligned} i D(\ln(u(x) + i) - \ln(u(x) - i)) &= i \left(\frac{u'(x)}{u(x) + i} - \frac{u'(x)}{u(x) - i} \right) \\ &= 2 \frac{u'(x)}{u(x)^2 + 1} = 2 D \arctan u(x). \end{aligned}$$

Sitzung 12.5 Man beachte, daß die beiden Funktionen $i(\ln(u(x) + i) - \ln(u(x) - i))$ und $2 \arctan u(x)$ zwar dieselbe Ableitung besitzen, sich aber um eine Konstante unterscheiden. Für $u \in \mathbb{R}$ ist die Differenz gleich π bei *Mathematicas* Wahl der Zweige für die Logarithmus- und Arkustangensfunktion. Für $u(x) = x$ haben wir beispielsweise

`In[1] := Plot[{I(Log[x + I] - Log[x - I]), 2 ArcTan[x]}, {x, -2, 2}]`

¹⁵ D bezeichne die Ableitung nach der Variablen x .



`Out[1]= -Graphics-`

Falls $u(x)$ reelle Pole hat, hat $\arctan u(x)$ Sprungstellen:

```
In[2]:= Plot[Evaluate[{I(Log[u + I] - Log[u - I]), 2ArcTan[u]}],
              u -> (x^3 - 2x)/(x^2 - 1)], {x, -2, 2}]
```

Wir berechnen die Differenz $2 \arctan u - i(\ln(u + i) - \ln(u - i))$:

```
In[3]:= simp = ComplexExpand[2ArcTan[u] - I(Log[u + I] - Log[u - I])]
```

```
Out[3]= -Arg(1 - i u) + Arg(i u + 1) - Arg(u - i) + Arg(u + i) +
         i (log(|1 - i u|) - log(|i u + 1|) + log(|u - i|) - log(|u + i|))
```

```
In[4]:= simp = FullSimplify[simp, u > -1 && u < 1]
```

```
Out[4]= -Arg(1 - i u) + Arg(i u + 1) - Arg(u - i) + Arg(u + i)
```

```
In[5]:= FullSimplify[Table[simp, {u, -3, 3}]]
```

```
Out[5]= {π, π, π, π, π, π, π}
```

Leider erkennt *Mathematica* nicht, daß dies konstant ist. □

Da wir nur an Integralen interessiert sind, kommt es auf die verschiedenen Integrationskonstanten aber nicht an. Allerdings müssen wir damit rechnen, daß in verschiedenen Intervallen verschiedene Integrationskonstanten gelten. Dies werden wir gleich betrachten.

Mit der Konversion (12.20) lassen sich alle Logarithmen mit rein imaginären Koeffizienten betrachten, welche paarweise konjugiert zueinander auftreten:

$$\begin{aligned}
 ci \ln(a + ib) - ci \ln(a - ib) &= ci \ln\left(\frac{a + ib}{a - ib}\right) \\
 &= ci \ln\left(\frac{\frac{a}{b} + i}{\frac{a}{b} - i}\right) = ci \ln\left(\frac{a}{b} + i\right) - ci \ln\left(\frac{a}{b} - i\right) \\
 &\equiv 2c \arctan \frac{a}{b}, \qquad (12.21)
 \end{aligned}$$

wobei wir mit $\equiv$ die Gleichheit modulo einer geeigneten Integrationskonstanten bezeichnen. Man beachte, daß bei der Anwendung der Logarithmenregel $\ln x - \ln y = \ln \frac{x}{y}$ wieder ein Wechsel des Zweiges der Logarithmusfunktion auftreten kann. Die Gleichung (12.21) gilt also ohnehin nur modulo π .

Sitzung 12.6 Als Beispiel berechnen wir

$$\int \frac{x^4 - 3x^2 + 6}{x^6 - 5x^4 + 5x^2 + 4} \cdot$$


Hierzu laden wir die Funktion `RothsteinTrager` aus Sitzung 12.4 und berechnen

```
In[1]:= RT = RothsteinTrager[r =  $\frac{x^4 - 3x^2 + 6}{x^6 - 5x^4 + 5x^2 + 4}$ , x]
```

```
Out[1]=  $\frac{1}{2} i \log(x^3 + i x^2 - 3 x - 2 i) - \frac{1}{2} i \log(x^3 - i x^2 - 3 x + 2 i)$ 
```

Wir stellen das resultierende Integral zunächst einmal graphisch dar:

```
In[2]:= Plot[RT, {x, -3, 3}]
```



```
Out[2]= -Graphics-
```

Wir wollen nun (12.21) anwenden. Leider hat *Mathematica* Schwierigkeiten mit dem Muster komplexer Zahlen. Beispielsweise erhalten wir¹⁶

```
In[3]:= Flatten[Cases[x3 + i x2 - 3 x - 2 i, A_ + I * B_ -> {A, B}, {0, infinity}]]
```

```
Out[3]= {x3 - 3 x - 2 i, x2}
```

auf diese einfache Weise kann man also nicht Real- und Imaginärteile komplexer Polynome abspalten.

Die – im Kopf – gemäß (12.21) konvertierte Funktion

¹⁶ Mit den Konstrukten `ComplexExpand[Re[...]]` und `ComplexExpand[Im[...]]` findet man allerdings den Real- und Imaginärteil.


$$\frac{i}{2} \ln(x^3 - 3x + i(x^2 - 2)) - \frac{i}{2} \ln(x^3 - 3x - i(x^2 - 2)) \equiv \arctan \frac{x^3 - 3x}{x^2 - 2}$$

hat folgenden Graphen:

```
In[4]:= int = ArcTan [  $\frac{x^3 - 3x}{x^2 - 2}$  ]
```

```
Out[4]=  $\tan^{-1}\left(\frac{x^3 - 3x}{x^2 - 2}\right)$ 
```

```
In[5]:= Plot[int, {x, -3, 3}]
```



```
Out[5]= -Graphics-
```


Beide Graphen stimmen wieder modulo π überein, aber nur in geeigneten Teilintervallen, welche durch die beiden Nullstellen $x = \pm\sqrt{2}$ des Nenners $x^2 - 2$ gegeben sind, und beide Integralfunktionen haben eine höchst unangenehme Eigenschaft: Sie sind *unstetig*, obwohl es auch eine stetige Stammfunktion gibt.¹⁷ Dies hat folgende Konsequenz: Wendet man den Hauptsatz der Differential- und Integralrechnung an, um das bestimmte Integral

$$\int_1^2 \frac{x^4 - 3x^2 + 6}{x^6 - 5x^4 + 5x^2 + 4}$$

zu bestimmen, welches aufgrund der Positivität des Integranden

```
In[6]:= Plot[r, {x, 1, 2}]
```

¹⁷ Man beachte allerdings, daß die Rothstein-Trager-Lösung am zweiten Pol $x = -\sqrt{2}$ *keinen* Sprung hat.



`Out[6]= -Graphics-`

positiv ist, so erhalten wir mit der Rothstein-Trager-Lösung

`In[7]:= bestimmt = (RT/.x → 2) - (RT/.x → 1)`

`Out[7]=` $-\frac{1}{2}i \log(-2-i) + \frac{1}{2}i \log(-2+i) - \frac{1}{2}i \log(2-2i) + \frac{1}{2}i \log(2+2i)$

`In[8]:= N[bestimmt]`

`Out[8]=` $-3.46334 + 0.i$

und mit der Arkustangens-Lösung

`In[9]:= bestimmt = (int/.x → 2) - (int/.x → 1)`

`Out[9]=` $\frac{\pi}{4} - \tan^{-1}(2)$

`In[10]:= N[bestimmt]`

`Out[10]=` -0.321751

in beiden Fällen also negative Werte, welche natürlich modulo π mit dem korrekten Integralwert übereinstimmen. Wegen der unstetigen Stammfunktion ist also eine Anwendung des Hauptsatzes unzulässig!

Welchen Wert liefert *Mathematica*s Integrationsprozedur? *Mathematica* berechnet¹⁸

`In[11]:= bestimmt =  $\int_1^2 r \, dx$`

¹⁸ Nach relativ großer Rechenzeit liefert *Mathematica* 4 das einfachere korrekte Resultat $\frac{5\pi}{4} - \arctan 2$.

$$\begin{aligned}
\text{Out}[11] = & \frac{1}{2} \left(3 \operatorname{RootSum} \left[\#1^6 - 5 \#1^4 + 5 \#1^2 + 4\&, \frac{\log(1 - \#1) \#1}{3 \#1^4 - 10 \#1^2 + 5} \& \right] - \right. \\
& 3 \operatorname{RootSum} \left[\#1^6 - 5 \#1^4 + 5 \#1^2 + 4\&, \frac{\log(2 - \#1) \#1}{3 \#1^4 - 10 \#1^2 + 5} \& \right] - \\
& \operatorname{RootSum} \left[\#1^6 - 5 \#1^4 + 5 \#1^2 + 4\&, \frac{\log(1 - \#1) \#1^3}{3 \#1^4 - 10 \#1^2 + 5} \& \right] + \\
& \operatorname{RootSum} \left[\#1^6 - 5 \#1^4 + 5 \#1^2 + 4\&, \frac{\log(2 - \#1) \#1^3}{3 \#1^4 - 10 \#1^2 + 5} \& \right] - \\
& 6 \operatorname{RootSum} \left[\#1^6 - 5 \#1^4 + 5 \#1^2 + 4\&, \frac{\log(1 - \#1)}{3 \#1^5 - 10 \#1^3 + 5 \#1} \& \right] + \\
& \left. 6 \operatorname{RootSum} \left[\#1^6 - 5 \#1^4 + 5 \#1^2 + 4\&, \frac{\log(2 - \#1)}{3 \#1^5 - 10 \#1^3 + 5 \#1} \& \right] \right)
\end{aligned}$$

In[12] := **N[bestimmt]**

Out[12] = 2.81984 + 0. *i*

offenbar den richtigen Wert, obwohl die zugehörige Stammfunktion

In[13] := **int** = $\int r \, dx$

$$\text{Out}[13] = \frac{1}{2} \tan^{-1} \left(\frac{x(x^2 - 3)}{x^2 - 2} \right) - \frac{1}{2} \tan^{-1} \left(\frac{x(x^2 - 3)}{2 - x^2} \right)$$

In[14] := **bestimmt** = (**int** /. **x** → 2) - (**int** /. **x** → 1)

$$\text{Out}[14] = \frac{\pi}{4} - \tan^{-1}(2)$$

In[15] := **N[bestimmt]**

Out[15] = -0.321751

hierfür nicht geeignet ist. □

Da man bestimmte Integrale lieber nicht nur modulo π bestimmen möchte, ist es also hilfreich, *stetige* Stammfunktionen zu finden, damit die Anwendung des Hauptsatzes der Differential- und Integralrechnung zulässig ist. Man sieht nun leicht ein, daß die Unstetigkeiten davon herrühren, daß die Argumente der Arkustangensfunktionen rationale Funktionen sind. An den Polen dieser rationalen Funktionen findet dann gegebenenfalls ein Zweigwechsel statt. Dies kann man vermeiden, wenn man die resultierende Stammfunktion als Summe von Arkustangensfunktionen schreibt, deren Argumente Polynome sind. Eine derartige Summe ist dann automatisch stetig. Dies wird mit dem folgenden Algorithmus von Rioboo [Rio1991] erreicht.

Satz 12.12 Sei $\mathbb{K}$ ein Körper mit $i \notin \mathbb{K}$. Seien weiter $a(x), b(x) \in \mathbb{K}[x]$ von 0 verschiedene Polynome. Seien schließlich $c(x), d(x) \in \mathbb{K}[x]$, $c(x) \neq 0$, derart, daß $b(x)d(x) - a(x)c(x) = g(x) = \gcd(a(x), b(x))$ sei, d. h., $g(x), c(x)$ und $d(x)$ findet man mit dem erweiterten Euklidischen Algorithmus.

Dann gelten die Gleichungen

$$\begin{aligned} Di \ln \frac{a(x) + ib(x)}{a(x) - ib(x)} &= Di \ln \frac{-b(x) + ia(x)}{-b(x) - ia(x)} \\ &= 2D \arctan \frac{a(x)d(x) + b(x)c(x)}{g(x)} + Di \ln \frac{d(x) + ic(x)}{d(x) - ic(x)}. \end{aligned} \quad (12.22)$$

Diese Gleichungen liefern einen rekursiven Algorithmus zur Darstellung von $i \ln(a(x) + ib(x)) - i \ln(a(x) - ib(x)) = i \ln \frac{a(x)+ib(x)}{a(x)-ib(x)}$ durch eine Summe von Arkustangentensternen mit polynomialen Argumenten.

Beweis: Die erste Gleichung folgt aus der Rechnung

```
In[1] := Together [
    D[Log[(a[x] + I b[x])/(a[x] - I b[x])] -
    Log[(-b[x] + I a[x])/(-b[x] - I a[x])],
    x]
Out[1] = 0
```

Sei nun also $g(x) := \gcd(a(x), b(x)) \in \mathbb{K}[x]$. Dann bestimmt man mit dem erweiterten Euklidischen Algorithmus $c(x), d(x) \in \mathbb{K}[x]$ mit $b(x)d(x) - a(x)c(x) = g(x)$. Wir schreiben in der Folge $p(x) := \frac{a(x)d(x) + b(x)c(x)}{g(x)}$. Da $g(x)$ sowohl $a(x)$ als auch $b(x)$ teilt, ist $p(x) \in \mathbb{K}[x]$. Wegen

$$\begin{aligned} \frac{a(x) + ib(x)}{a(x) - ib(x)} &= \left(\frac{d(x) - ic(x)}{d(x) + ic(x)} \cdot \frac{a(x) + ib(x)}{a(x) - ib(x)} \right) \cdot \frac{d(x) + ic(x)}{d(x) - ic(x)} \\ &= \left(\frac{a(x)d(x) + b(x)c(x) + i(b(x)d(x) - a(x)c(x))}{a(x)d(x) + b(x)c(x) - i(b(x)d(x) - a(x)c(x))} \right) \cdot \frac{d(x) + ic(x)}{d(x) - ic(x)} \\ &= \left(\frac{a(x)d(x) + b(x)c(x) + ig(x)}{a(x)d(x) + b(x)c(x) - ig(x)} \right) \cdot \frac{d(x) + ic(x)}{d(x) - ic(x)} \\ &= \frac{p(x) + i}{p(x) - i} \cdot \frac{d(x) + ic(x)}{d(x) - ic(x)} \end{aligned}$$

und mit (12.20) ist

$$Di \ln \frac{a(x) + ib(x)}{a(x) - ib(x)} = 2D \arctan p(x) + Di \ln \frac{d(x) + ic(x)}{d(x) - ic(x)},$$

also die zweite Gleichung von (12.22).

Der rekursive Algorithmus zur Bestimmung einer stetigen Stammfunktion funktioniert nun wie folgt.

Falls $\deg(a(x), x) < \deg(b(x), x)$ ist, kann man mit der ersten Gleichung aus (12.22) die Rolle von $a(x)$ und $b(x)$ vertauschen. Wir können also o. B. d. A. annehmen, daß $\deg(b(x), x) \leq \deg(a(x), x)$ ist.

Ist nun $b(x) \mid a(x)$, so verwenden wir die Transformation (12.21)

$$D i \ln \frac{a(x) + i b(x)}{a(x) - i b(x)} = 2D \arctan \frac{a(x)}{b(x)}. \quad (12.23)$$

Dies ist bereits eine Darstellung durch einen Arkustangensterm mit Polynomargument.

Wir nehmen nun in der Folge an, daß $b(x)$ kein Teiler von $a(x)$ ist. Mit dem erweiterten Euklidischen Algorithmus bestimmen wir dann den größten gemeinsamen Teiler $g(x) = \gcd(a(x), b(x))$ sowie $c(x), d(x) \in \mathbb{K}[x]$ derart, daß $b(x)d(x) - a(x)c(x) = g(x)$ gilt. Hierbei ist $\deg(d(x), x) < \deg(a(x), x)$, s. Übungsaufgabe 12.2. Die zweite Gleichung in (12.22) liefert dann also die rekursive Vorschrift

$$D i \ln \frac{a(x) + i b(x)}{a(x) - i b(x)} = 2D \arctan p(x) + D i \frac{d(x) + i c(x)}{d(x) - i c(x)}$$

mit einem (weiteren) Arkustangenssummanden mit Polynomargument. Wegen $\deg(d(x), x) < \deg(a(x), x)$ ist sichergestellt, daß die Rekursion terminiert. $\square$

Sitzung 12.7 Für den erweiterten Euklidischen Algorithmus laden wir

```
In[1] := Needs["Algebra'PolynomialExtendedGCD']
```

und programmieren den Rioboo-Algorithmus wie folgt:

```
In[2] := Clear[LogToArcTan]
```

```
LogToArcTan[a_, b_, x_] :=
  LogToArcTan[-b, a, x] /; Exponent[a, x] < Exponent[b, x]

LogToArcTan[a_, b_, x_] :=
  2 ArcTan[a/b] /; PolynomialQ[Together[a/b], x]


LogToArcTan[a_, b_, x_] := Module[{g, c, d},
  {g, {d, c}} = PolynomialExtendedGCD[b, -a];
  2 ArcTan[Together[(a * d + b * c)/g]] + LogToArcTan[d, c, x]
]
```

Für unsere Beispielfunktion aus Sitzung 12.6 erhalten wir

```
In[3] := rioboo = LogToArcTan[x^3 - 3x, x^2 - 2, x]
Out[3] = 2 tan-1(x) + 2 tan-1(x^3) + 2 tan-1(1/2 (x^5 - 3x^3 + x))
```

welches eine stetige Darstellung als Summe von Arkustangenstermen mit polynomialen Argumenten ist:

```
In[4]:= Plot[ $\frac{\text{rioboo}}{2}$ , {x, -3, 3}]
```



```
Out[4]= -Graphics-
```

Man sieht, daß sich nun auch automatisch eine weitere wünschenswerte Eigenschaft der Stammfunktion ergeben hat: Die zu integrierende Funktion $\frac{x^4-3x^2+6}{x^6-5x^4+5x^2+4}$ ist eine gerade Funktion, also gibt es eine ungerade Stammfunktion. Der Rioboo-Algorithmus liefert diese. $\square$

12.5 Ergänzende Bemerkungen

Die Kombination der betrachteten Algorithmen liefert einen vollständigen Algorithmus zur rationalen Integration. Bis auf einige effizienzsteigernde Adaptionen ist dies der augenblickliche Stand der Dinge auf diesem Gebiet, und wir haben gesehen, daß dieser Algorithmus beispielsweise in *Mathematica* bislang nicht vollständig eingebaut ist.¹⁹

Beim Risch-Algorithmus zur Integration elementarer Funktionen sind algebraische, logarithmische und exponentielle Differential-Körpererweiterungen für Eingabe und Ausgabe erlaubt. Wie im rationalen Fall liefert ein Strukturtheorem eine Aussage darüber, wie eine mögliche Stammfunktion aussehen kann: Zur Bildung der Stammfunktion sind nur logarithmische Körpererweiterungen nötig. Allerdings treten kompliziertere Türme von Körpererweiterungen auf und die Theorie wird algebraisch anspruchsvoller. Letzten Endes kann man die Stammfunktion wieder durch eine Art Koeffizientenvergleich bestimmen. Als Literatur zu diesem Thema sind die Bücher [GCL1992] und [Bro1997] zu empfehlen.

¹⁹ Diese Bemerkung gilt (mindestens) bis *Mathematica* Version 5.0

ÜBUNGSAUFGABEN

12.1 Zeigen Sie, daß alle möglichen Differentialoperatoren in $\mathbb{Q}(x)$ durch die beiden Bedingungen $D(1) = 0$ und $D(x) = F(x)$ gegeben sind, wobei $F(x) \in \mathbb{Q}(x)$ eine beliebige rationale Funktion ist.

12.2 Zeigen Sie: Seien $a(x), b(x) \in \mathbb{K}[x]$ und sind $g(x) = \gcd(a(x), b(x))$ sowie $c(x), d(x) \in \mathbb{K}[x]$ die Polynome, welche vom erweiterten euklidischen Algorithmus erzeugt werden, so daß $b(x)d(x) + a(x)c(x) = g(x)$ gilt, dann ist $\deg(d(x), x) < \deg(a(x), x)$.

Zeigen Sie an einem Beispiel, daß die Gleichung $b(x)d(x) + a(x)c(x) = g(x)$ auch von Polynomen $c(x), d(x) \in \mathbb{K}[x]$ erfüllt werden kann, die die Beziehung $\deg(d(x), x) < \deg(a(x), x)$ verletzen.

12.3 Man zeige: Hat ein reelles Polynom $a(x) \in \mathbb{R}[x]$ eine komplexe Nullstelle $z_0 \in \mathbb{C}$, so ist auch die konjugiert komplexe Zahl $\overline{z_0}$ eine Nullstelle von $a(x)$.

12.4 Finden Sie heraus, für welche $a \in \mathbb{N}$, $a \leq 1000$ das Polynom $x^4 + a \in \mathbb{Z}[x]$ eine echte ganzzahlige Faktorisierung und damit die rationale Funktion $\frac{1}{x^4+a} \in \mathbb{Q}(x)$ eine echte rationale Partialbruchzerlegung besitzt.

12.5 Vergleichen Sie die Resultate für $\int \frac{1}{x^4+4}$

(a) von Mathematica;

(b) des Rothstein-Trager-Algorithmus;

(c) durch Anwendung einer rationalen Partialbruchzerlegung des Integranden.

12.6 Geben Sie zwei weitere – möglichst einfache – Beispiele für den Rioboo-Algorithmus.

12.7 Im der Analysis-Vorlesung wird häufig die $\tan \frac{x}{2}$ -Substitution vorgestellt, mit welcher man jede trigonometrische rationale Funktion $R(x) \in \mathbb{R}(\sin x, \cos x)$ integrieren kann, da die Substitution auf einen rationalen Integranden führt. Zeigen Sie an einem Beispiel, daß diese Substitution im allgemeinen ebenfalls zu unstetigen Stammfunktionen führt. Wird dies vom Rioboo-Algorithmus behoben?